

了解 Excel	4
Excel 特性	4
Excel 模式	5
为什么学 Excel	5
持续学习必要性	5
你学习 Excel 目的是	6
Excel 用户级别	6
课程内容	7
基本知识	7
(1) 工作簿属性	7
(2) 工作表属性	7
(3) 单元格属性	8
Excel 中的数据类型	8
Excel 基本操作技巧	8
1. 使用定位条件方法选择特区域	9
2. 选择性粘贴功能	10
3. 使用查找方法选中特定格式区域	11
4. 组合键功能	13
5. 利用填充柄功能提高效率	14
6. 设置单元格式化, 区别编辑栏与数据的不同	15
7. 设置工作簿属性与保护功能, 增强安全性	15
8. 打印设置	17
9. 拆分和冻结	17
数据收集与清洗	18
1. 快速输入数据	19
1. 利用自定义列表	19
2. 填充柄	19
3. 填充柄+快捷菜单	20
4. 多个不连续单元格同数据	20
5. 文本记忆式输入法	20
6. 数据验证方法	20
7. 数字自定义格式	21
8. 函数输入(10-99)	21
9. 填充对号或叉号	21
9. 同时填充多个工作表	21
11. 生成一万个序号	21
10. VBA 代码填充数据	22
2. 导入(获取外部数据)	22
3. 使用快速填充功能提取信息	22
从省市列表中提取出市或省	22
合并市省	23
提取日期中的年	23
从不规则的数字与字母组合中提取出数字	23
4. 文本型数字转换为数值文本型数字	23

4. 自定义数据格式.....	24
1.把数值以数值+文本的形式显示:	24
2.把文本以文本+其它文本的方式显示	24
3.在某一区域中突出显示某种类型的数字.....	25
4.设置格式:1 表示对勾 0 表示为叉号	25
不足 5 位用 0 补齐.....	26
批量删除单引号	26
5. 规范化数据	26
1.不规范日期 1: 以英文句号分割的文本型日期.....	27
2.不规范日期 2: 年代以 2 位数表示的文本型日期.....	27
2.不规范日期 3: 以 8 位数字表示的文本型日期	27
3.分列: 固定宽度/日期格式	28
合并日期和时间	28
6. 分列与合并计算.....	28
分列-源数据	29
使用分列对歌名进行提取.....	29
使用分列对不规范的 20140101 类型的日期数据进行规范化	30
合并计算	30
9 批量复制照片.....	31
7. 确保正输入数据方法.....	32
公式.....	32
公式计算 核心地位.....	32
1. 公式功能.....	33
2. 重新计算公式的时间和方式.....	33
3. 公式组成.....	33
公式中的 5 种常量.....	34
4. 公式运算符	34
5. 单元格地址引用	35
1 单元格地址引用	36
99 乘法表	36
本利和.....	37
单列变多列-地址引用.....	37
多列转换为一列	38
课后作业 1-文本连接&	39
课后作业 2-单元格引用巧妙.....	39
输入, 编辑复制公式.....	39
函数公式出错信息 (7 个).....	39
函数错误种类.....	40
错误值的利用	41
不同的工作表计算	42
跨工作表的计算	42
跨工作簿的计算	43
公式追踪	43
追踪公式引用	43

追踪错误	44
圈释无效数据	45
显示所有的公式与隐藏公式	45
逐步理解函数的实现	46
F9 功能	47
公式分类	47
课后作业 1-类型转换案例	47
课后作业 2-数据整理实例 1	48
函数	48
函数	48
函数难点	49
函数输入方法	49
函数工具与技巧	50
Excel 常用函数	50
数学函数	50
统计函数	53
文本函数	59
逻辑函数	64
日期函数	65
生成间断序号	73
引用与查找函数	74
CHOOSE 函数	74
FIND 函数	75
Offset()	76
MATCH	82
INDEX 函数	86
VLOOKUP	95
行, 列标号函数	102
INDIRECT 函数	102
初级水平用户	107
图片链接	107
数组	108
什么是数组?	108
数组大小要求	108
数组公式	108
数组公式标志	109
数组维数	109
规律	109
数组优点	110
运算符*和+ 代替逻辑函数 AND 和 OR	110
在数组中不能用 AND, OR 代替 *, +	110
视频演示	111
计算多种商品的销售总金额	111
两列数据之差的和	111

判断身份证的位数是否合法.....	112
多条件求和: 统计一车间男职工工资总和。	113
判断男女是否达到退休年龄.....	114
SUMPRODUCT 统计 3 天同品牌销售之和。	115
产量大于 100 的和	116
SMALL 和 LARGE 函数.....	118
数组函数 FREQUENCY	119
课后作业 1-平均分高于 90 的少数民族	119
课后作业 2-前三名产量和.....	120
课后作业 3-姓赵	121
进价-多条件求和	121
图表.....	123
数据透视表.....	125
数据透视表.....	125
1. 数据透视表.....	125
2. 创建数据透视表.....	126
3. 值显示方式.....	126
4. 组合功能	126
5. 数据透视表计算.....	127
6. 多个工作表数据.....	127
7. 切片器	127

了解 Excel

Excel 特性

数据

高级计算器

加减乘除计算的电子表格

无关联, 零散数据

文字

日期时间

图片

信息

科学管理决策的工具

精细化量分析的工具

建立数据, 工作表簿逻辑关系

报表呈现, 图表呈现

数据分析模型，系统

Excel 模式

数据

规范准确存储数据

高效率统计数据

呈现数据相互关系

关联与整合数据

提取与查找数据

寻求数据发展趋势以及规律

决策的支持以及参考

信息

强大!

复杂!

精准!

为什么学 Excel

大数据时代，数据量大，从手工到 自动化

灵活，便捷，高效率

获取计算机处理问题的思维

学习数据处理的 思路与方法

感悟 EXCEL 设计理念

学习 EXCEL 处理问题的逻辑思路

积累职场高效率处理数据的 技巧

持续学习必要性

Excel 中，有那么一些 强大而简单的功能，如果没有人告 诉，可能你永远都不会知道。

通过百度搜索， 需要知道其关键字，才能获取。

问题往不是解决了，大多数时候我们 还不知道 竟然可以"这样"处理,"解决

多数时候，解决问题 缺乏的是思路和方法，缺乏的是 积 累和思考，缺乏的是 站在巨人的肩上

你学习 Excel 目的是

- (1) 应用性 学习:出于应用而学习 , 学以致用 ;
- (2) 守株性学习:技多不压身 ,越多好 越多好 ,学了再说 ,等待某天派上大用场;
- (3) 娱乐性学习 :出于兴趣而学习, 愉悦;
- (4) 逻辑性 学习 :为训练逻辑能力而学习;
- (5) 献身性学习 献身性学习 :出于帮助他人而学习 ;
- (6) 空白性学习 空白性学习 :出于填补空白时间而学习 ,实在是闲得没事干了;
- (7) 提升性 学习: 更加深入, 掌握其精髓;
- (8) 挑战性 学习: 勇于进取, 难不倒拓宽 EXCEL 技巧与知识面。

Excel 用户级别

新手 :

基本操作方法和常用功能 :

初级用户 :

建立表格, 图化 :

中级用户 :

理解并熟练各菜单命令,

使用数据透视表,

掌握 20 余个函数,

嵌套余个函数,

嵌套宏,

名称宏,

名称条件格式,

有效验证,

组及分级显示高筛选,

自定义排序

高级用户 :

熟练运用数组,

引与查找函,

动态图表,

动态数据透视表, 分析工具以及 VBA

专家 :

高超的技术并拥有丰富行业知识和经验

课程内容

Excel 数据处理技巧和高效方法
Excel 数据收集, 和清洗
Excel 公式, 函数 , 数组及数据分析
Excel 可视化数据分析(排序, 筛选, 透视表等)
Excel 图表化数据呈现
Excel 宏与交互式界面 设计
Power BI

基本知识

(1) 工作簿属性

工作簿 , 即 Excel 文档 , 扩展名 `xlsx` 及其他类型

设置文档属性

重点 掌握其中的 , 主题 , 关键词 , 作者

设置 密码 :

"另存为 "对话框 /工具 /常规选项

(2) 工作表属性

工作表缺省用户名 (`sheet1`, `sheet2`, `sheet3` 等)

工作表列: `A, B, ..., XFD` ($16384=2^{14}$)

工作表行: `1, 2, ..., 1048576` $= 2^{20}$

工作表单元格数目: $2^{14} \times 2^{20}$

工作表操作

插入 /删除

复制 /移动

重命名

保护

设置工作表标签颜色

(3) 单元格属性

单元格 地址： 列标行, A1 ,B23

区域地址：

模式：左上角单元格:右下角

A1:D4

D2:D2000

1:4

A:A

A:C

单元格所在行, 列的高度 和宽度

选定单元格或 区域

数据列: Ctrl+Shift+向下箭头

数据 行: Ctrl+Shift+向右箭头

数据 多行列: Ctrl+Shift+向右箭头/向下箭头

中间不能有空白单元格, 否则就不能选中所有的区域.

Excel 中的数据类型

数据类型示例

文本姓名, 姓别, 住址, 职务, 职称, 符号等

数值 0, 1, 2, 3, 4, ..., 日期, 整数, 小数等都是数字类型的

逻辑值 TRUE, FALSE

错误值 #VALUE!, #DIV/0!, #NAME!, #N/A, #REF!, #NULL!

数字 1 格式化为日期为 1900/1/1

日期格式化为整数, 为 1900/1/1 到现在的天数.

时间格式化为小数

先格式化单元格为文本, 再输入数字就是文本型数字

Excel 基本操作技巧

问题 1: 总销售额

公式 > 自动求和

问题 2: 统计地区销售额, 含有条件的求和. 把地区复制到新列中, 选中包含标题行在内的

所有单元格, 数据 > 数据工具 > 删除重复项

使用下拉框 选中单元各 > 数据 > 数据工具 > 数据验证 > 序列 > 来源:上面删除重复项后的除标题在内的所有单元格

=sumif(range, criteria, [sum_range]), 第一个参数选择除标题外的没有删除重复项的所有销售地区列表, 第二个参数选择设置数据验证的单元格, 第 3 个参数选择除标题之外的所有销售额.

问题 3: 销售额在 2000-5000 之间的数据之和

sumifs 多条件求和

sumifs(sum_range, criteria_range1, criterial1, [criteria_range2, criterial2])

第 1,2,4 个参数选择除标题在外的所有销售额, 第 3 个参数为条件, 设置为">2000", 第 5 个参数为"<5000". 表示在销售额中对 2000-5000 之间的数据求和.

问题 4: 一班男同学获奖次数 sumproduct 条件求和之王

各个条件之间是与的关系

sumproduct((班级列="一班")*(性别列="男")*获奖次数列)

问题 5:统计单元格个数

数字个数 count(数据区域)

空单元格个数 countblank(数据区域)

非空单元格个数 counta(数据区域)

使用 SUMPRODUCT 函数多条件求和, 语法为:

=SUMPRODUCT((区域 1=条件 1)*(区域 2=条件 2)* 求和区域)

1. 使用定位条件方法选择特区域

知识点:

使用"定位条件"选中某特的单元格

(1)特定条件: 空值

输入 100, 按 Ctrl+Enter

(2)特定条件: 公式产生的错误值

视频演示

1. 使用定位条件选择空单元格

选中单元格区域 > 查找和选择或 F5 定位 > 定位条件 goto special > 空值

2. 批量修改单元格

方法: 先使用定位条件选择区域中空单元格, 输入 100, 按组合键 **Ctrl+Enter**
在编辑栏输入 0 > **ctrl+Enter** 把所有空单元格都变成 0

3. 使用定位条件复制分类汇总的结果

默认的复制粘贴会得到原数据, 想要不复制原数据的内容, 只复制分类汇总的结果
选中数据区域 > 定位条件 > 可见单元格 > 复制 > 粘贴
可以用到分类汇总的结果, 筛选的结果.

4. 使用定位条件选中错误单元格

由公式产生的错误, 无法使用函数计算.

选中单元格 > 定位条件 > 公式 > 只保留"错误" > **delete** 删除或 清除 > 清除内容

2. 选择性粘贴功能

转置

加减乘除运算

复制部分设置

跳过空单元格

视频演示

1. 选择性粘贴-运算

某一列所有数据增加 25%

某个单元格中填入 1.25 > 复制 1.25 所在单元格 > 定位到要修改金额的第一个单元格中 >
ctrl+shift+向下箭头选中所有的单元格 > 选择性粘贴 > 运算 > 乘
数字格式 > 会计专用 > 格式化为货币

2. 选择性粘贴, 将两列数据合为一列

辅助列. 输入 1, 第 2 个为空, 选中这两个单元格, 向下拖动到原数据行的结尾处. 生成一列
1, 空单元格, 2, 空单元格, 3, 空单元格, 4, 空单元格, 5...这样一列包括标题在内的与原来两

列数据平行的

选中这三列 > 定位条件 > 常量, 这样就跳过了空白单元格

开始 > 单元格 > 插入 > 插入工作表行 > 所有数据中都间隔一行插入了一行空白行 > 复制要插入的源数据, 即金额行, 并且跳过金额行中第 1 个空白单元格, 从有金额的一行开始向下复制. > 定位到要插入的目标行即部分名称行要插入的第一个位置处 > 粘贴 > 选择性粘贴 > 跳过空单元格

3. 选择性粘贴-数据转换乘

文本型数据不能直接进行运算, 可以转换为数值型数据进行运算. 还可以使用选择性粘贴来完成.

在任一个单元格中输入 1 > 复制此单元格 > 选中所有的文本型数据 > 选择性粘贴 > 乘 > 所有文本型的数据就会转换为数字.

3. 使用查找方法选中特定格式区域

知识点:

(1)使用"查找"选中某特定格式的单元

方法: 从 单元格中选择 格式

(2)单元格匹配

视频演示

1. 查找功能统计同填充色个数

开始 > 数字 > 对话框启动器 > 设置单元格格式 > 数字, 对齐, 字体, 边框, 填充 4 个选项卡用来设置单元格格式

2. 取消边框

开始 > 字体 > 下框线 > 无框线

选择同格式的所有单元格

选中某一个单元格 > ctrl+F 查找 > 格式右下拉箭头 > 从单元格选择格式 > 点击要选择的单元格 > 查找全部 > 在查找与替换中选中所有的结果 > 关闭查找替换的窗口, 同格式的所有单元格仍处于选中状态 > 设置单元格格式

再次使用查找功能时一定要清除之前设置的格式

清除单元格格式

选择单元格区域 > 开始选项卡 > 编辑区域 > 清除 > 清除格式

3. 单元格匹配替换

知识点: 将成绩为 0 的替换为"补考".

查找与替换 > 格式右下拉箭头 > 清除查找格式 > 查找内容: 0 > 替换为: 补考 > 全部替换 > 发现 60 中的 0 也被替换掉 > 打开查找与替换中的选项 > 选中"单元格匹配" > 全部替换

4. 替换单元格中的*号

方法: 查找为"~*"

5. 一列变 10 列

A1:A100 中为 100 个名字, 想要在 C1:L10 这 10*10 个单元格中逐行输入这 100 个名字.

方法:

1. 在 C1 输入 A1, 填充到单元格 L1, 得到 A1, A2, ... A10;
2. 在 C2 输入 A11, 填充到单元格 L2, 得到 A11, A12, ... A20;
3. 选中这 2 行, 复制到 L10, 得到 A21-A30, A31-A40, A41-A50, ... A91-A100;
4. 选中 C1:L10 这 10*10 的单元格区域, 使用查找替换把 "A" 替换为 "=A", 这样, 单元格中就变成了公式, 也就是单元格的引用了, 就可以把 A1:A100 中的内容分散到 10*10 的区域中.

因为是用公式产生的, 如果删除 A1:A100 中的名字, 得到的 C1:L10 中的内容也会被清空, 选择 C1:L10, 复制 > 定位到一个新的单元格 > 选择性粘贴 > 数值, 只粘贴公式的结果, 得到的内容是文本了, 不再是使用公式产生的了.

6. 数据验证-多行变 1 行

A1:G11 中是 77 个名字, 想把名字放在一个下拉列表中

Step1: 公式 > 定义的名称 > 名称管理器 > 删除本工作表定义的名称

Step2: 定位到某一单元格 > 数据 > 数据工具 > 数据验证 > 允许:序列 > 来源:选中多行多列的单元格区域 > 确定会出错 > "列表源必须是划定分界后的数据列表, 或是对单一行或单一列的引用", 要先定义一个名称

Step3: 公式 > 定义的名称 > 定义名称, 名称相当于数学中的变量 > 名称:姓名, 范围:工作簿, 引用位置:先删除原有内容, 再选中一行或一列 > 确定

定位到要生成下拉列表的单元格中 > 数据 > 数据工具 > 数据验证 > 允许: 序列 > 来源:=姓名, 等号就是公式 > 确定, 现在就在下拉列表中生成了选择的一行或一列中的姓名

公式 > 定义的名称 > 名称管理器 > 修改姓名 > 引用位置: 先删除原有的内容, 再选中多行多列 > 确定 > 现在就会根据多行多列生成序列.

注意: 如果在一开始就设置名称的来源为多行多列, 就会出错. 所以先使用一列或一行的, 再通过修改名称把名称的来源改为多行多列

4. 组合键功能

1 <Ctrl> 键 (多重选取)

2 <Shift> 键 (首尾范围选取)

3 <F8>键(扩展模式)

4 Ctrl+Shift +方向键

5 <Shift+Home>键 --边界扩展--->行首

6 <Ctrl+Shift+Home>键--边界扩展--->表首

7<Ctrl+Shift+END>键--边界扩展--->最后一个使用中的 最后一个使用中的 最后一个使用中的 单元格

8 <Ctrl+A>键 (当前区域选取)

9 <ALT+;>键 (选取可见单元格)

11 使用 F5 键定位功能(条件定位)

12 ALT+= , Ctrl+Enter

Ctrl+; 在当前单元格中填入今天的日期

视频演示

1. 组合键 ALT+=汇总上方数据

ctrl 选中所有需要得到和值的空白单元格 > ALT+=完成小计

或者

选中需要得出结果的所有单元格区域 > 定位条件 > 空值 选中所有的空白单元格 > alt+=

2. 组合键功能，快速选中大数据区域

对于连续的数据，没有空白行列或单元格

数据列: ctrl+shift+向下箭头

数据行: ctrl+shift+向右箭头

数据多行多列: ctrl+shift+向下箭头 > 向右箭头

3. 快速选定不连续的单元格

按下组合键"Shift+F8"，激活"添加选定"模式，此时工作表下方的状态栏中会显示出"添加到所选内容"字样，以后分别单击不连续的单元格或单元格区域即可选定，而不必按住 Ctrl 键不放。

5. 利用填充柄功能提高效率

拖曳

双击

批量输入数据

视频演示

1. 填充柄复制公式

选中 E2 单元格，将鼠标置于右下角，光标显示为黑色十字，双击填充柄"黑色十字"，复制一列同类型公式

公式中引用的单元格中不能有空白行，列或单元格

2. 辅助列-隔行插入 1 空行

辅助列+序列+排序

插入一列名为辅助列，除标题行外的第一个单元格中输入 1，填充柄拖动都是 1，ctrl+填充柄拖动就会成为 1, 2, 3..., 列数与原数据相同.

在辅助列列尾填入 1.5，按着 ctrl 再拖动与原数据相同长度的数据 > 填充进 2.5, 3.5, 4.5 ... 的一列数据

右击辅助列中任一单元格 > 排序 > 升序 > 原数据每一列下都插入了一个空白行. 使用辅助列的排序带动所有行内容的排序，从而生成空白行

如何恢复原数据 > 车次 > 排序 > 升序

插入两行三行如何实现呢?

3. 辅助列-提取工资条的标题

需要把工资条的标题行复制到每一个数据行中.

插入辅助列 > 填充与原数据列等长的 1, 2, 3...值 > 复制标题行到最下方 > 由于标题行已经存在,只需要再填充比原数据列小 1 的标题行即可 > 用 1.4,2.4,3.4...填充与新复制标题行等长的辅助列 > 因为空白列要插入到每一个标题-数据行的下面 > 在最下面空白行中填充 1.2,2.2,3.2...的数据 > 点击辅助列 > 排序

6. 设置单元格格式，区别编辑栏与数据的不同

单元格格式：

数字(典型)

对齐

字体

填充

边框

7. 设置工作簿属性与保护功能，增强安全性

使用密码来保护工作簿

另存为 > 工具 > 常规选项 > 打开权限密码 > 修改权限密码

审阅 | 更改

保护工作表

保护工作簿

部分单元格编辑保护

视频演示

1. 工作簿保护

结构, 表示对工作表的编辑限制;
设置工作簿属性与保护功能, 以提高文档的安全性

审阅 > 更改 > 保护工作簿 > 输入密码

此时不能删除, 新建, 重命名, 修改或删除工作表, 但是可以对工作表的内容进行编辑. 所以保护工作簿实际上是对它其中下一级对象的编辑的限制. 但由于工作表的内容是工作簿的下下一级内容, 保护工作簿并不能限制对工作表内容的修改.

2. 工作表保护

对单元格编辑的限制;
保护工作表某些单元格不能修改或编辑。

审阅 > 更改 > 保护工作表 > 选择需要保护的内容 > 允许此工作表的所有用户进行: 选中 "选定锁定单元格"和"选定未锁定单元格", 其它一律不勾选 > 输入密码 > 此时就无法在单元格中输入或修改内容, 也不能插入行插入列或删除行删除列. 如果在"允许此工作表的所有用户进行"中 同时勾选了"插入行"或"插入列", 那么在保护工作表之后, 也能够进行插入行或插入列的操作.

3. 部分单元格编辑

工作表中某些内容可以编辑, 有些内容则不允许编辑, 如在收集用户信息并从 excel 中导入到数据库中时, 只有用户自己的信息允许修改, 其它的都不允许修改.

方法: 取消可以编辑单元格的"锁定", 然后设置保护工作表

选中可以编辑的单元格 > 开始 > 数字 > 设置单元格格式 > 保护 > 取消勾选"锁定" > 审阅 > 更改 > 保护工作表 > 输入密码 > 只有取消锁定的单元格才可以编辑, 其它的都不能编辑.

也可以使用隐藏功能来隐藏单元格中的公式.

8. 打印设置

顶端标题

网格线, 默认没有网格线, 需要手动设置

所有数据的工作表

视频演示

每页打印标题

设置第 6 行为顶端标题行, 第 2 页首行是源数据第 6 行值.

页面布局 > 页面设置对话框启动器 > 工作表选项卡 > 顶端标题行 > 选择标题行为第 6 行 > 打印预览, 第 2 页, 第 3 页都会出现标题

同样, 如果横向的内容超出来一页, 也可以设置左端标题行

9. 拆分和冻结

冻结首行

视图 > 窗口 > 冻结窗格 > 冻结首行

拆分为上下两个窗格

选中要拆分的某行 > 视图 > 窗口 > 拆分

冻结拆分的窗格

定位到拆分后的某个窗格中 > 视图 > 窗口 > 冻结窗格 > 冻结窗格

拆分为四个窗格

定位到要拆分的行中某单元格中 > 视图 > 窗口 > 拆分

课后作业 1-插入 3 行空行

使用辅助列 1-26 > 1.1-26.1 > 1.2-26.2 > 1.3-26.3 > 排序

课后作业 2-工资条

要求:每一行数据都加上标题,同时在数据下面加一行空格

方法:

1.选中从第 2 行到第 24 行共 23 > 插入行 > 插入 23 行空行 > 复制标题行(共 24 行) > 对 24 行标题行添加辅助列 > 数据 1-24

2.在原数据行的辅助列填充 1-24,

3.在原数据行下方的辅助列生成 24 行空白行,填充 1.1-24.1

4.升序排序

课后作业 3-选中条件格式

作业: 选中设置了条件格式的单元格,清除条件格式。

选中所有的单元格 > 定位条件 > 条件格式 > 全部 > 清除 > 清除格式

课后作业 4-行内容的区别

作业:选中两列中每行内容不同的单元格

选中所有的单元格 > 定位条件 > 内容差异单元格

课后作业 5-快速删除文本框

定位 > 对象 > 清除 > 全部清除或 del

课后作业 6-跳过空单元格

作业: 两列互补的数据合并为一列

选择性粘贴的"跳过空单元格"方法应用。

方法:

复制其中一行数据到新的列 > 选中另一互补列对应的区域 > 复制 > 定位到新列对应位置 > 选择性粘贴 > 跳过空单元格

数据收集与清洗

清洗数据,整理数据,把不规范的数字规范化

批量输入数据方法

批量输入图片
导入数据
快速填充
数据类型转换
数据自定义格式
规范化数据
分列
合并计算

1. 快速输入数据

- (1)自定义数据序列
- (2)填充柄(带文本与数字混合, Ctrl 辅助, 等比差序列日期)

系统提供的数据序列
(3)填充柄 +快捷菜单
(4)多个不连续单元格输入相同内容(也可)Ctrl+Enter
(5)文本记忆式(字符型, 快捷菜单的"选择列表")
(6)数据验证方法
(7)数字自定义格式
(8)函数方法
(9)同时填充多个工作表
(10) VBA 代码填充数据
(11)批量输入图片

视频演示

1. 利用自定义列表

Excel 选项 > 高级 > 常规 > 编辑自定义列表 > 新增序列 > a, b, c, d, e, f > 在某一单元格中输入 c > 填充柄拖动 > 就会循环出现刚才定义的序列.

甲星期一 1 月

2.填充柄

文字+数字等差序列 QW001 > 填充柄拖动 > QW002, QW003, ...
数字 输入 10001 > ctrl+拖动填充柄
等差数列 2, 4 > 拖动填充柄

等差数列的应用: 函数 $z=x^2+y^2$, 生成二次曲面

x:-5,-4.5,-3.5...4.5,5 > 复制 x 的数据 > 选择性粘贴 > 转置 > 粘贴到 y 中

3.填充柄+快捷菜单

等比序列

输入第一个值 > 鼠标右键拖动 > 序列 > 类型: 等比序列 > 步长值: 3

工作日填充

第一个单元格中填入当前日期: ctrl+; 在编辑栏中看到是原始的数据, 而单元格中显示的则是经过格式化修饰过的数据, 格式化为日期即可在单元格中看到年月日的日期, 也可以使用格式刷 > 鼠标右键拖动 > 序列 > 日期单位 > 工作日

数字 > 设置单元格格式 > 数字 > 日期 > 类型: 3 月 14 日

右键拖动 > 以月填充, 以年填充

4.多个不连续单元格同数据

ctrl 选中或者 Shift+F8 选中 > 输入 100 > ctrl+Enter

5.文本记忆式输入法

性别 系别

男 经济系

女 法律系

定位到每一行下面的空白单元格 > 右键 > 从下拉列表中选择

6. 数据验证方法

选中需要进行下拉选择的单元格或区域 > 数据 > 数据验证 > 序列 > 来源:男,女

把数据验证单元格复制到其它地方

复制单元格 > 选中目标单元格或单元格区域 > 选择性粘贴 > 粘贴:验证

用户调查时,可以使用下拉列表选择的方式,选择项一般要保持在 10 个以内.

7.数字自定义格式

选中所有的目标单元格或单元格区域 > 数字 > 设置单元格格式 > 分类:自定义 > 类型: [=1]"男";[=0]"女" > 选中性别为男的单元格 > 输入 1 > ctrl+Enter > 变为男 > 选中性别为女的单元格 > 输入 0 > ctrl+Enter > 变为女

8.函数输入(10-99)

随机数

RAND(), 生成 0,1 之间的随机数

ROUND(), 生成带小数位的数.

=ROUND(RAND()*100,2)

=ROUND(RAND(),2)

10-99 之间的数字

=INT(10+90*RAND())

或者=RANDBETWEEN(10, 99)

9.填充对号或叉号

选中单元格区域 > 选择字体 wingdings2 > 数据 > 数据验证 > 设置 > 允许:序列 > 来源: R,S > R 就是带框对号, S 就是带框叉号

9.同时填充多个工作表

ctrl 或 shift 同时选中多个工作表 > 在当前激活的工作表中输入内容并设置格式 > 所有工作表中的内容都会改变

11. 生成一万个序号

在 A 列生成 1 至 10000 的连续序号。

方法一:

在 A1 单元格中输入 1 > 名称框中输入 A1:A10000, Enter, 选中了 A1:A10000 的单元格区域

开始 > 编辑 > 填充 > 序列 > 类型:等差数列, 步长:1, 终止值:10000, 序列产生在:列

方法二:

在 A1 单元格中输入 1 > 开始 > 编辑 > 填充 > 序列 > 序列产生在:列, 等差数列, 步长:1, 终止值:10000

10. VBA 代码填充数据

2. 导入(获取外部数据)

导入数据 Access

导入文本 等

定位到要导入的单元格 > 数据 > 从文本&csv > 选择文本文件 > 加载到 > 选择工作表

定位到要导入的单元格 > 数据 > 从文本&csv > 选择文本文件 > 加载到 > 数据的放置位置: 现有工作表 > 点击某空白单元格 >

选择 access 数据库的时候可以选择一次性导入多个工作表

3. 使用快速填充功能提取信息

从不规范高校名称数据中提取出规范的高校名称 > 输入第一个高校名称 > 右键拖动填充柄 > 快速填充

输入第一个高校的编号 > 右键拖动填充柄 > 快速填充 > 会把高校编号的数据提取出来.

提取数据中的数字部分

第一个单元格中输入原单元格中的数字部分, 右键拖动, 快速填充

从省市列表中提取出市或省

在第一个单元格中输入对应的省市列表中的市 > 右键拖动填充柄 > 快速填充 > 自动填充市信息

在第一个单元格中输入对应的省市列表中的省 > 右键拖动填充柄 > 快速填充 > 自动填充省信息

但是多于两个字的省市信息都不能正常填充

合并市省

省列表和市列表右边一列的第一个单元格中输入沈阳辽宁 > 定位到下一个单元格 > Ctrl+E
自动填充前两列中的数据

提取日期中的年

输入对应的年份 > 右键拖动 > 快速填充

从不规则的数字与字母组合中提取出数字

对字母和数字组合进行排序

A122, A29, A43, A7, B20, B243, C234

右边平行空白单元格中输入 A > 右键拖动填充柄 > 填充字母和数字组合中的字母

右边平行空白单元格中输入 122 > 右键拖动填充柄 > 填充字母和数字组合中的数字

对数字列进行升序排序 > 对字母列进行升序排序 > 原字母和数字组合列得到按字母和数字组合进行排序的结果

4. 文本型数字转换为数值文本型数字

- 点击 "智能标记", 选择"转换为数字"

例: 统计姓陈的员工数

=SUMPRODUCT(--(LEFT(A2:A10)="陈"))

作业: 使用公式将 A 列数据转换到 B 列, 且为数值型数据。

方法: 使用 6 个公式之一转换, 若结果没变化, 将其单元格格式设置为"常规"。

=A1*1

=A1/1

=A1+0

=A1-0

=--A1 减负运算(第 1 个是减法, 第 2 个是负数)

=VALUE(A1)

操作: 定位到 B 列第二行, 在编辑栏中输入=A1*1, Enter > B2 就转换为数字了 > 拖动填充柄

4. 自定义数据格式

组成一：四个部分：

正数格式;负数格式;零;文本

组成二：

大于条件值;小于条件值;等于条件值;文本

方法：

实例 1: 数值 +文本: 0"/年"

实例 2: 文本 +文本: @"年级 "

实例 3: [红色][<=100];[蓝色][>100]

实例 4: [=1]"√";[=0]"X"

视频演示: 自定义数据格式

1.把数值以数值+文本的形式显示:

要求:

在单元格中显示 50 元/年, 但在编辑栏中则显示为 50, 可以用于工龄工资的计算上

在目标单元格中输入数字 > 开始 > 数字 > 设置单元格格式 > 数字 > 分类:自定义 > 类型:0"元/年"

其中的 0 表示原码, 用来保存原始单元格数据的有效位, 什么是无效位, 代表实际值的位数, 如 5.01 的有效位是 3, 而 005.01000 中前面 2 个 0 和后面 3 个 0 都不是有效位.

2.把文本以文本+其它文本的方式显示

可以用于给公司的部门添加公司的名称

第 1 步:

在目标单元格中输入公式, 等于前一列的数据

第 2 步, 方法 1:

开始 > 数字 对话框选择器 > 设置单元格格式对话框 > 数字选项卡 > 分类:自定义 > 类型:@"年级"或"年级"@

第 2 步, 方法 2:

开始 > 设置单元格格式 > 自定义 > 类型:@"年级"

0 代表数字, @代表文本

3.在某一区域中突出显示某种类型的数字

正数, 负数和零分别用不同颜色的数字表示

要求:

设置格式: 正数-默认色; 负数-红色+括号; 零-蓝色, 且输入的数字都都转化为对应的百分比值.

操作:

选中某一区域 > 数字 > 设置单元格格式 > 分类:自定义 > 类型:0.0%;[红色](0.0%);[蓝色]0.0 > 在这些区域中输入数字, 就会看到效果

说明:

0.0%;[红色](0.0%);[蓝色]0.0 中

0.0 为原码, 表示保留一位小数

0.0%;表示把正数转换为带 1 位小数的百分比值.

[红色](0.0%);表示 把负数转换为红色的带括号的带 1 位小数的百分比值..

[蓝色]0.0 表示把 0 转换为蓝色的带 1 位小数的值

说明:

Excel 中的数据格式:

<正数>;<负数>;<零>;<文本>

数字格式最多可包含四个代码部分,各个部分用分号分隔。这些代码部分按先后顺序定义正数, 负数, 零值和文本的格式。

Excel 只有两种数据类型,一是文本,一是数字, 把日期, 时间都归为数字类中.

把某个单元格格式化为日期, 输入 1, 因为 Excel 可表示的最小的日期为 1900 年 1 月 1 日, 输入的 1 就会变为 1900/1/1.

把单元格格式化为时间, 输入一个接近 1 的数字, 就会被格式化为接近 24 点的时间.

4.设置格式:1 表示对勾 0 表示为叉号

自定义格式:[=1]"√";[=0]"×"

插入符号方法

这种方法在 Word 中同样适用, 选择 Excel 插入菜单, 然后选择"特殊符号", 这时会出现"插入特殊符号"窗口, 选择"数学符号"选项卡, 找到"√"这个对号符号, 按确定即可

Alt+符号法

这种方法在很多地方可以适用,也就是按住 Alt 键,再输入对号√对应的字母,这个对号√符号就出来了,对号√对应的字母数字是 41420。

带框对号符号怎么打

在 Excel 单元格中输入大字 R,然后选择字体中的"wingdings2"这种字体,这时就会出现带框对号符号了,这种方法其实就是选择 R 的图文字体

不足 5 位用 0 补齐

设置数字自定义格式: 00000

批量删除单引号

输入 3/4 会自动转换为日期, 输入 5:6 会自动转换为时间.

把数字转换为文本型数字的方法:

选中单元格 > 数字 > 数字格式下拉框 > 文本 >

或者以单引号为开头, 输入数字

要批量删除某区域中所有单元格中文本的单引号。

方法:

1. 格式刷复制一个空单元格。

2. 复制到这个区域即可。

或者使用 ctrl+shift+C 复制一个空白单元格格式, ctrl+shift+V 把这个格式粘贴到目标单元格

5. 规范化数据

(1)日期格式

=SUBSTITUTE(A2,".","/")*1

=--TEXT(A5,"#-00-00")

(2)时间格式

=--TEXT(A10,"00!:00!:00")

(3)分列, 提取货币数字

数据 |数据工具 |分列

(4)整理数据方法

step1: 在"百度知道" 搜索降央卓玛歌曲名称大全

step2: 复制整个文本到 Excel

step3: 分列, 隔符号是 ";";然后转置

Step4: 删除数字方法: 复制到 Word, 使用特殊符号的"任意数字"为查找, 替换为空

step5: 再复制到 Excel

视频演示

1.不规范日期 1：以英文句号分割的文本型日期.

windows 中设置的格式决定了 Excel 中日期的格式

方法一:

选中日期所在单元格 > 查找与替换 > 查找内容 . 替换为 / > 设置单元格格式 > 日期 > 选择其它格式

方法二:使用函数

定位到新的平行单元格 > 编辑框输入 = SUBSTITUTE(> 选择要替换的单元格 > =SUBSTITUTE(B3,".","/") > 双击填充柄把公式复制到此列下面所有的单元格中完成余下数据的替换

得到的是文本型的日期, 无法转换为其它的日期格式, 需要转换为数字型的日期, 使用 =SUBSTITUTE(B3,".","/")*1 来转变为数字型的日期

选中某个日期单元格 > 数字 > 数字格式下拉框 > 常规 > 转换为数字, 表示从 1900 年 1 月 1 日到此日期的天数

2.不规范日期 2：年代以 2 位数表示的文本型日期

使用 TEXT 函数转换为文本型日期

TEXT 函数通过加日期的分隔符-的方法把原来的 6 位文本型数字表示的日期转换为 6 位 excel 可以识别的日期.

=--TEXT(F3,"#-00-00")或=--TEXT(F3,"00-00-00")

TEXT(F3,"#-00-00")表示从右向左开始取数, 取最右边的 2 位文本型数字, 在它们前面加上-., 取倒数 3,4 位的文本型数字, 在它们前面加上-, 这样就转换 6 位的为文本型日期. 再使用—TEXT 转换为数字型日期

2.不规范日期 3：以 8 位数字表示的文本型日期

=--TEXT(F3, "0000-00-00")

从右边取 4 位, 2 位 2 位表示日和月, 然后把左边余下的 4 位数格式化为 4 位的年.

3.分列：固定宽度/日期格式

规范时间，没有使用冒号来分隔的时间

=--TEXT(A10,"00!:00!:00")

因为在 Excel 中冒号：有特殊的意义，即表示选中的单元格的区域，要把冒号显示时间的冒号，必须加一个强字符！

选中时间的单元格 > 数字 > 数字格式 > 常规 > 就会把时间转换为小数表示的数字

注释：要在 excel 中显示冒号；，必须加一个强字符！或-

合并日期和时间

日期转换为数字格式是以整数显示的，时间转换为数字格式是以小数显示的，可以直接把这两种数据相加，就得到了日期和时间

停车日期列，停车时间列，定位到合并日期和时间列中，=停车日期列+停车时间列 > 自定义单元格格式，选择日期中的日期+时间的显示样式

拆分日期和时间(日期)，取出整数部分=INT(合并的日期和时间)

拆分日期和时间(时间)= 合并的日期和时间列 > 格式化为时间

计算间隔多少小时 = (离开日期列+离开时间列-(停车日期+停车时间))*24

计算间隔多少分钟 = (离开日期列+离开时间列-(停车日期+停车时间))*24*60

6. 分列与合并计算

分列： "数据"选项卡的分列命令

清除不可见字符

利用同样的方法，还可以清除掉数据中不可见字符。如果从系统导出看起来正常，格式也是规范但无法求和汇总不妨试用分列的方法处理一下。

转换 8 位日期

8 位数值，列据格式选择 **【日期】**，在下拉框中选择 **【YMD】**。

清除身份证号码中的空格

在弹出的**【文本分列向导】**对话框中选择**【固定宽度】**

提取身份证号码中的出生日期：建立分列线

合并计算："数据"选项卡的"合并计算"命令

视频演示

分列-源数据

原数据列: 中国 VS 阿尔及利亚

要求: 提取不重复参赛国家名。

知识点: 数据分列。

选中原数据列 > 数据 > 数据工具 > 分列 > 文本分列向导 > 分隔符号 > 下一步 > 但是分隔符号中没有字母, 只有符号, 所以要先要把数据进行替换

选中所有的原数据 > 查找与替换 > 把 VS 替换为, > 全部替换 > 分列

在生成的两列中插入一个空白列, 这样对第 1 列的操作就不会影响到第 2 列
给生成的第一列添加"国家"的标题 > 定位到第一列中任意一个单元格 > 数据 > 数据工具 > 删除重复项 > 勾选"数据包含标题", 会自动选择国家的字段

删除重复项第二种方法: 数据透视表。

对于生成的第二列数据 > 添加标题:国家 > 选中包括标题在内的所有数据或者只定位到某个单元格 > 插入 > 数据透视表, 就会自动选择单元格所在的所有列中的数据 > 创建数据透视表 > 选择数据透视表的位置 > 现有工作表 > 选择新一行的一个单元格 > 打开的数据透视表字段右工具栏, 把字段列表中的国家拖动到行和值中 > 就会生成包括国家和国家出现次数的表

复制数据透视表中生成的国家, 粘贴到第一列删除重复项后的国家后面 > 再次对添加新的国家后的第一列执行删除重复项的操作 > 数据 > 数据工具 > 删除重复项

使用分列对歌名进行提取

01.我的母亲 02.金瓶似的小山 03.请喝一杯下马酒 04.爱在天地间 01. 东女国 02. 吉祥的酥油灯 03. 月亮升起来 04. 美女天仙(藏语)05. 牧人 06. 鸿雁 07. 索呀啦(藏语)08. 雪域名城 09. 慈母颂(藏语)10. 这山这水 11. 故乡情(藏语)12. 黑帐篷 05.草原夜色美 06.一剪梅 07.再唱山歌给党听 08.国家 09.驼铃 10.远飞的大雁 11.阿瓦古丽 12.边关有我 01.草原上升起不落的太阳 02.走天涯 03.美丽的草原我的家 04.多情总为无情伤 05.阿尔斯楞的眼睛 06.敖包相会 07.这片草原 08.姑娘我爱你 09.草原之夜 10.故乡 11.泪在投降 12.慈祥的母亲 13.天堂的草原 01.西海情歌 02.我和草原有个约定 03.呼伦贝尔大草原 04.蓝色的蒙古高原 05.陪你一起看草原 06.草原恋 07.牧人 08.卓玛 09.父亲的草原母亲的河 10.天边 11.故乡 12.手心里的温柔 13.雨中飘荡的回忆

知识点: 导入网页数据到 Excel。

方法:

step1:在百度知道查找"降央卓玛歌曲名称大全"

step2: 复制整个文本到 Excel

step3: 分列,分隔符号是"." > 选择性粘贴转置到列中

step4: 消除数字 > 复制到 Word > 查找与替换 > 更多 > 定位到查找 > 特殊符号 > 任意数字^# > 替换为空

消除数字方法二: 使用快速填充, 在平行单元格中输入除编号外的歌名, 拖动填充柄就可以提取出所有的歌名, 但是对于有"(藏语)"的歌名, 只会删除掉后面的括号.

step5: 复制到 Excel

step6: 删除空格 定位到右边平行列中相应的第一个单元格 > =TRIM() 点击左边单元格 > 双击填充柄完成余下数据的填充

使用分列对不规范的 20140101 类型的日期数据进行规范化

选中除标题行的所有的日期 > 数据 > 分列 > 文本分列向导 > 固定宽度 > 添加分列线 > 列数据格式: 常规或日期 > 选择目标区域

生成的年月日分别在不同的单元格中 > 使用函数把它们合并到一个单元格中 > 定位到右边相邻的平行单元格中 > =DATE(年列,月列,日列), 分别用鼠标点选选择年月日 > 生成新的数据

选中所有的日期 > 数据 > 分列 > 文本分列向导 > 固定宽度 > 添加分列线中不添加任何分列线 > 列数据格式: 日期 YMD > 选择目标区域 > 直接就能生成合并的日期了
使用这种方法也可以直接把 6 位的文本型的数字日期转换为 Excel 可以识别的日期

在"导入 text 数据"工作表中, 运费之平均值为用文字表示的人民币金额, 无法直接进行运算, 可以使用分列转换为数字

运费 之 平均值
¥200.05
¥5.00
¥5.00
¥50.00
¥4.00
¥7.00
¥7.00
¥12.00
¥10.00

选中所有的运费的值 > 数据 > 分列 > 文本分列向导 > 固定宽度 > 添加分列线 > 完成 > 把金额数据拖动到原数据处 > 开始 > 数字 > 数字格式 > 会计专用

合并计算

已知上半年和下半年的数据, 合并得到全年的数据

分 4 种情况:

1. 源数据最左列标题与首行标题顺序相同.
2. 源数据仅首行标题顺序相同.
3. 源数据仅最左列标题顺序相同.
4. 源数据最左列标题与首行标题顺序均不同.

第 1 种情况:

必须首先保证上半年和下半年数据的格式和结构完全一样.

由于源数据与目标数据的左侧标题和上标题顺序完全相同, 只需要选择数据即可.

选中全年中除标题的所有数据的单元格 > 数据 > 合并计算 > 函数:求和 > 引用位置:选择上半年工作表中除标题外的数据 > 添加 > 选择下半年工作表中除标题外的数据 > 添加 > 如果上半年与下半年店面的名称顺序不一致,就要勾选首行 > 如果产品的顺序不一致,就会勾选最左列 > 取消二者的勾选 > 确定

如果上半年和下半年中左列产品标题中的顺序不一致, 在计算时全年的表中就不能有左列的标题, 先要删除全年的表中左列的产品标题, 计算时要选择包含左列标题列在内的单元格, 同时勾选求和中的最左列. 每个店的数据也是类似的.

Excel 中对列排序很方便, 如果对行排序, 要根据店面的名称排序 > 选中除产品列在内的所有单元格 > 数据 > 排序 > 选项 > 排序选项 > 方向:按行排序 > 方法:字母排序 > 在排序对话框中, 行区域,主要关键字: 行 1, 这里也不会显示出行标题 > 再次进行上面的合并操作, 这时全年的表中就不能有左侧标题和上面的标题, 先要删除左和上的标题, 再选择全年的所有单元格区域, 在计算时选择上半年和下半年工作表中包括行标题和列标题的单元格区域, 并且勾选"首行"和"最左列".

如果数据太多, 无法确定是否顺序相同, 就选择所有的区域, 并且选中首行和最左列, 这样就不会出错.

9 批量复制照片

作业: 批量输入照片。

要把员工的照片放在员工的信息工作表中

其中 A2 单元格中是员工的姓名, 注意照片是以员工姓名为文件名的照片,才能正确导入.

复制文件夹的路径, 在 windows 中按 shift, 右键中选择复制为路径

1.在 H2 单元格输入公式:

= "<table>", 并复制到 H10 单元格, 这时上面公式中的"&A2&"就会被替换为员工的姓名

2.然后选中 H2: H10,复制, 粘贴到记事本中, 借助记事本把内容变成文本, 会删除等号和一些引号, 然后再复制所有数据到 H2: H10;

3.选中所有有照片的行 > 鼠标拖动两行的分割线批量调整行高: 165,列宽 23.(这里的宽高与

照片的分辨率数据不是一个单位);

如果第一次没有成功, 再做时会有很多小方框, 选中照片所在的行, 使用定位条件, 选择对象, 删除即可.

7. 确保正输入数据方法

设置单元格输入范围

数据验证

统一格式

自动更正

下拉列表选择数据

公式

单元格地址引用

不同工作表, 簿之间的计算

显示和打印公式的方法

追踪公式的引用单元格, 从属错误

利用公式求值和 F9 功能键逐步解析公式的方法

分析公式中出现的 7 种错误原因

案例

公式计算 核心地位

Excel 的众多功能中, 公式计算位于核心地内置函数与强大 的公式计算能力, 确保 Excel 成为世界上最普及的数据运算工具

功能

计算准确, 速度 快捷

10 秒之内计算结果

修改联动, 自动更新

区别手动与自效果

操作简单, 轻松 易学

一个公式解决一问题

1. 公式功能

Data-->数据录入-->数据整理-->计算与分析-->呈现与表达-->Information

主要作用:

1. 计算

2. 建立数据之间的关联

单元格数据之间无关系, 各工作表之间无关系, 各工作簿之间无关系

通过公式的逻辑关系, 把它们联起来

核心: 确立数据之间的关联关系。实现的是一种算法, 通过其结果来描述这种关系。

2. 重新计算公式的时间和方式

公式 > 计算 > 计算选项 > 自动, 手动

自动重新计算(默认的设置)的情况

1. 只有在公式所依赖的单元格发生更改

2. 第一次打开工作簿以及编辑时

3. "选项" 的"公式"类别中的"计算选项"部分的"工作簿计算"下, 单击"自动"

除数据表外, 自动重算

1. 若要在每次更改值, 公式或名称时重新计算除数据表之外所有相关的

手动计算

1. 若要关闭自动重新计算

2. 单击"手动"时, Excel 将自动选中"保存工作簿前重新计算"复框。如果存工作簿需要很长时间, 那么清除"保存重新计算"可缩短。

如果表内的公式不多, 就使用自动, 如果公式或数据非常多, 刷新需要很长时间, 就使用手动计算。

3. 公式组成

=PI()*A2^2

公式的组成部分

1. 函数: PI() 函数返回值 pi:3.142...

2. 引用: A2 返回单元格 A2 中的值

3. 常量: 2 直接输入公式中的数字或文本值

4. 运算符: ^ (脱字号) 运算符表示将数字乘方, *(星号)运算符表示相乘

函数

预先编写的程序, 作用: 把分散在各个单元格,各个工作表,各个工作簿的数据有机的关联起来.

单元格的引用方式

单元格的相对地址, 绝对地址, 混合地址, 同一工作簿其他工作表中的单元格, 不同工作簿中工作表的单元格, 名称

常量

直接输入公式中的数字或文本值

运算符

是一个标记或符号, 指定表达式内执行运算的类型。有数学, 比较文本连接和引用运算符

公式中的 5 种常量

- a) 数值常量, 如 `=(5+8)*3`
- b) 日期常量, 如 `=DATEDIF("2012-1-1",NOW(),"m")`
- c) 文本常量, 如 `= "国际"&"复合"`
- d) 逻辑值常量, 如 `=VLOOKUP("张三丰",A:B,2,FALSE)`
- e) 错误值常量, 如 `=COUNTIF(A:A, #DIV/0!)`

4. 公式运算符

1. 算术运算符

`+`, `-`, `*`, `/`, 负号, `%`, `^`

2. 关系运算符:

`>`, `<`, `>=`, `<=`, `=`, `<>`

3. 文本运算符 :

`&` 不是逻辑的和.

4. 引用运算符:

`:`

`,`

空格(函数参中使用)

优先级: 引用运算符 > 算术运算符 > 文本运算符 > 关系运算符

算术运算符	含义	示例
<code>+(加号)</code>	加	<code>3+3</code>

-(减号)	减	
*(星号)	乘	3*3
/ (斜杠)	除	3 月 3 日
%(百分号)	百分号	20%
^(脱字符)	乘方	3^2

关系运算符

=(等号)	等于	A1=B1
>(大于号)	大于	A1>B1
<(小于号)	小于	A1<B1
>=(大于等于号)	大于等于	A1>=B1
<=(小于等于号)	小于等于	A1<=B1
<>不等于	不等于	A1<>B1

运算符的优先级

序号	符号	类型	说明
1	: 空格 ,	引用运算符	冒号, 单个空格, 逗号
2	-	算术运算符	负号
3	%	算术运算符	百分号
4	^	算术运算符	乘幂
5	* /	算术运算符	乘 除
6	+ -	算术运算符	加 减
7	&	文本运算符	连接文本
8	= < > <= >= <>	比较运算符	比较两个值

5. 单元格地址引用

- (1) 相对引用
例： B6, A4, C5:F8
- (2) 绝对引用
例： \$B\$6, \$A\$4, \$C\$5:\$F\$8
- (3) 混合引用
混合引用： 例 B\$6,A\$4, C\$5:F\$8, \$B6, \$A4, \$C5:\$F8
- (4) F4 功能——在 4 种引用之间切换

使用引用时不要手动去输入单元格的地址，而是使用鼠标点选，让函数自动生成引用地址.

如果公式要复制给其它单元格，引用的行和列地址就会发生变化，就要考虑绝对引用和相对

引用的问题.

视频演示

1 单元格地址引用

知识点: 公式地址引用。

作业: 使用公式计算 H, J 列, 在 3 行输入一个公式, 其余双击填充柄复制。

技巧: 哪个不变就锁定哪个, 在哪个前面加\$

	A	B	C	D	E	F	G	H
1								20%
2	员工编号	员工姓名	所属部门	工作职位	手机号码	入职日期	合同月薪	本次调薪
3	AD801	钱明远	人事	主管	0531	2005/7/1	¥ 5,000	
4	AD802	孙小艾	财务	经理	2315	2005/9/1	¥ 10,000	
5	AD803	王铭国	财务	职员	3323	2003/12/5	¥ 4,000	
6	AD804	许晓群	行政	职员	0422	2005/2/8	¥ 4,500	
7	AD805	严品枝	财务	职员	8532	1995/3/5	¥ 4,000	
8	AD806	张建文	人事	职员	8852	2004/5/2	¥ 3,500	
9	AD807	王嘉义	销售	经理	7288	2007/1/2	¥ 12,000	
10	AD808	赵新乐	销售	主管	2286	1995/6/6	¥ 7,000	
11	AD809	何友华	行政	职员	2353	2005/1/3	¥ 5,000	
12	AD810	方丽丽	销售	职员	0231	1999/9/8	¥ 6,000	

把所有人的工资增加 20%

在 H1 中输入 20% > 定位到 H3, 输入=H3*(1+H1) > 双击填充柄完成下面的操作 > 如果公式|计算|计算选项中选择为手动, 则所有数据都与第一个数据相同, 如果选择为自动, 则 H4=G4*(1+H2), H5=G5*(1+H3), ... 增加的量并不是全都取的 H1 单元格中的数据, 而是依次引用 H1, H2, H3, ... > 想要固定引用 H1, 就要使用绝对引用 > 定位到编辑栏中的 H1 > 按 F4

=G3*(1+\$H\$1)表示行列数据都不变

=G3*(1+H\$1)表示行为绝对引用不会变化而列为相对引用可以变化

=G3*(1+\$H1)表示列为绝对引用而行为相对引用

> 对 H1 使用绝对引用\$H\$1, 对 G3 使用混合引用, \$G3 列不变为绝对引用, 行变化相对引用 > 双击填充柄完成余下数据的计算

99 乘法表

	A	B	C	D	E	F	G	H	I	J
1		1	2	3	4	5	6	7	8	9
2	1	1	2	3	4	5	6	7	8	9
3	2	2	4	6	8	10	12	14	16	18
4	3	3	6	9	12	15	18	21	24	27
5	4	4	8	12	16	20	24	28	32	36
6	5	5	10	15	20	25	30	35	40	45
7	6	6	12	18	24	30	36	42	48	54
8	7	7	14	21	28	35	42	49	56	63
9	8	8	16	24	32	40	48	56	64	72
10	9	9	18	27	36	45	54	63	72	81

行 B1-BJ 和列 A2-A10 中分别输入 1, 2, 3, ...9> 在 B2 中输入=A2*B1

以第 2 行为例, 向右拖动填充柄时行号不变列号变, 但根据乘法表的要求, 在第 2 行中拖动时第 1 个因子要固定在 A2 单元格中, 即列号不能发生变化, 所以锁定列\$A2.

以第 2 列为例, 向下手动填充柄时列号不变行号变, 但根据乘法表的要求, 在第 2 列中拖动时第 2 个因子要固定在 B2 单元格中, 即行号不能发生变化, 所以锁定行 B\$2.

Excel 中可以保存的最大的数为 10^{308}

列方向上下拖动填充柄时, 列号不变行号变, 一般固定行号

行方向左右拖动填充柄时, 行号不变列号变, 一般固定列号

本利和

知识点:单元格地址引用----文本公式转换为 excel 的公式。

本利和 = 本金 \times (1 + 年利率 \times 存期年数)

	A	B	C	D	E
1	本利和				
2	存期 (年)	1	2	3	5
	年利率				
3	本金	3.00%	3.50%	4.00%	4.25%
4	10000				
5	20000				
6	30000				
7	40000				
8	100000				

定位到 1 年的本利和单元格 B4 中 $\geq A4*(1+B3*B2) \geq$ 拖动填充柄到所有行和列

开始 > 单元格 > 格式 > 自动调整列宽

本金向下拖动时行变列不变, 固定列. 年利率和存期向右拖动时列变行不变, 固定行.

定位到 1 年的本利和单元格中 > 编辑栏中使用 F4 进行绝对引用和相对引用的切换 > $=\$A4*(1+B\$3*B\$2)$

单列变多列-地址引用

要求: 快速将单列数据转换成多列。

1 列 100 个姓名变成 5 列, 20 个/列。

复制到每个单元格，利用地址引用变化。

方法:

在 B1 单元格输入=A21 > 向右拖动填充柄到 E1, 包括 A 列共 5 列,> 双击 E1 的填充柄向下填充 > A1:E20 中的数据就是想要的结果

第 1 行中,A1, =A21, =A41, =A61, =A81

A 列中 A1:A100,B 列中 A21- A100, C 列中 A41- A100, D 列中 A61- A100, E 列中 A81-A100

多列转换为一列

要求: 快速将多列数据转换成单列。

方法:

1.在最左列输入公式有效数据的第 2 列地址, 例如, =B1,

2.复制到右第 5 列宽, 直到出现 0 为止, 向下拖动到第 100 行,此时最左列是所需数据

视频演示

A1:E20 中有 100 个名字, 把这 100 个名字转换到一列中

	A	B	C	D	E
1	赵一	何二一	云四一	鬱六一	乐八一
2	钱二	吕二二	苏四二	鲍六二	于八二
3	孙三	施二三	潘四三	史六三	时八三
4	王四	华二四	郎四四	薛六四	康八四
5	周五	孔二五	奚四五	费六五	皮八五
6	吴六	曹二六	范四六	廉六六	卞八六
7	郑七	严二七	彭四七	岑六七	齐八七
8	卫八	姜二八	马四八	汤六八	卜八八
9	冯九	金二九	鲁四九	雷六九	伍八九
10	陈一〇	魏三〇	韦五〇	贺七〇	余九〇
11	褚一一	陶三一	昌五一	倪七一	元九一
12	杨一二	喻三二	万五二	毕七二	黄九二
13	蒋一三	戚三三	苗五三	滕七三	顾九三
14	沈一四	谢三四	凤五四	殷七四	孟九四
15	韩一五	邹三五	花五五	罗七五	平九五
16	许一六	童三六	柳五六	常七六	尹九六
17	朱一七	柏三七	俞五七	郝七七	和九七
18	姜一八	水三八	任五八	郇七八	穆九八
19	尤一九	窦三九	索五九	安七九	萧九九
20	张二〇	葛四〇	唐六〇	傅八〇	汪一〇〇
21	何二一	云四一	鬱六一	乐八一	0

在 A21 单元格中输入 =B1 > 拖动 A21 单元格的填充柄到 E21 > 再拖动 E21 单元格的填充柄到 E100 > A1-A100 中就是原来的 100 个名字

第二种方法, 定位到 A21 单元格中 > 输入=B1 > 在名称框中输入 A21:E100, 回车选中 A21-E100 的区域 > ctrl + D 填充

但是此时如果清除源单元格中的数据,最左边的数据也被删除了 > 选择所有的数据 > 复制 > 选择性粘贴 > 在原来第一个单元格中粘贴:数值 > 公式就消失了, 就可以把多余的数据删除掉了

课后作业 1-文本连接&

知识点:文本连接运算符&使用。

作业:使用&运算符合并姓与名列的数据。

定位到姓名列第一行单元格中 > 输入=> 点击 B2> 输入&> 点击 C2> 回车 > 双击填充柄完成余下数据的填充 > 选择包含标题的姓名列的数据 > 复制 > 开始 > 粘贴 > 粘贴值 > 就可以删除原数据了 > 选中姓名所在的整列 > 剪切 > 在 B 列右键中选择插入剪切的单元格

课后作业 2-单元格引用巧妙

将 A,B,C...Z 一系列的数据放在一个单元格中.

定位到 B2> 输入=B1&A1> 双击填充柄

题目:将歌名合并并在 1 个单元格内。

www.excelhome.net

输入，编辑复制公式

手工输入公式

在公式中选择单元格引用

将区域 名称 粘贴到公式中

=G3*增长率

向公式中插入函数

使用"函数库"组

使用"插入函数"对话框

使用公式记忆键入

使用嵌套函数

函数公式出错信息 (7 个)

错误类型	产生原因	原因举例
#DIV/0!	零作除数	

#NAME?	在公式中使用了不能识别的名称	删除了公式中使用的名称, 或者不存在. 函数名的拼写错误
#VALUE	使用了不正确的参数或运算符	在需要数字或逻辑值时输入了文本
#REF!	引用了无效的单元格地址	删除了由其它公式引用的单元格 将移动单元格粘贴到由其它公式引用的中.
#NULL!	指定了两个并不相交的区域	使用了不正确的区域运算符或单元格引.
#N/A	当在函数或公式中引用了无法使的值	内部函数或自定义工作表中缺少一个多参. 使用的自定义工作表函数不存在. VLOOKUP() 函数中的查找值 lookup_value, FALSE/TRUE 参数指定了不正确的值域.
#NUM!	数字类型不正确	在需要数字参的函中使用了不能接受的参数. 由公式产生的数字太大或小: 数字必须在-10^307 和 10^307 之间
##### !	不属于错误(列宽设置问题)	输入到单元格中的数值太长, 在单元中显示不下 公式所产生的结果太长, 单元格容纳不下 日期和时间产生了负值.

视频演示

函数错误种类

知识点:了解公式与函数出错种类及原因。

数据	数据	显示公式	错误类型	原因:
TRUE	YY	=AVERAGE(A2:B2)	#DIV/0!	求平均函数的区域非数值, 产生除数为 0 的错误
		=单价*数量	#NAME?	没有定义"单价", "数量"这 2 个名称, 产生名称错误
	acd	=SQRT(A5)	#VALUE!	开平方函数的参数应该是数值, 而非文本, 产生值错误
	#REF!	=A9+B8	#REF!	公式=A9+B8 复制到左侧, 产生了引用错误, 由于 A 是最小的列号, 往左移没有更小的列号了
		=1*10^390	#NUM!	公式产生的数字太大超出 Excel 范围
1	3			
2	4	=SUM(A9:A10 B9:B10)	#NULL!	A9:A10 与 B9:B10 两个数据区域的交集是空集, 产生空串, 空格即取两个区域的交集

#N/A 查找替换时未查找到时出现的错误

	A	B	C
1	五年高考统计		

2	年份	平均总分	升学率
3	2005 年	570.22	0.9053
4	2006 年	568.04	0.9123
5	2007 年	562.38	0.9453
6	2008 年	564.32	0.9389
7	2009 年	553.23	0.9545

2004 年精确的升学率	
=VLOOKUP("2004 年", A3:C7, 3, FALSE)	#N/A
平均总分不超过 564 分的升学率	
=VLOOKUP(564, B3:C7, 2, TRUE)	0.9453
=VLOOKUP(564, B3:C7, 2, FALSE)	#N/A

错误值的利用

知识点:利用错误功能获取价值。(了解, 不做)
对同行的两组数据进行比对, 找出不同的数据.

C5	:	✕	✓	<i>f_x</i>	=MATCH(B5,\$A\$2:\$A\$26,0)
	A	B	C	D	
1	省直辖市自治	省直辖市自治区	找不同数据		
2	辽宁省	辽宁省	1		
3	吉林省	吉林省	2		
4	黑龙江省	黑龙江省	3		
5	上海市	上海	#N/A		
6	江苏省	江苏省	5		
7	山东省	山东省	6		
8	河南省	河南省	7		
9	湖北省	湖北省	8		
10	湖南省	湖南省	9		
11	广东省	广东	#N/A		
12	广西省	广西省	11		
13	海南省	海南省	12		

统计养老金大于 1000 元的年份:

C2								=COUNT(0/(\$B\$2:\$B\$14>=1000))
	A	B	C	D	E	F		
1	年度	北京市基本养老金(元/月)	大于1000元的年份					
2	2014年	3050	1					
3	2013年	2773	1					
4	2012年	2510	1					
5	2011年	2268	1					
6	2010年	2032	1					
7	2009年	1830	1					
8	2008年	1580	1					
9	2007年	1377	1					
10	2006年	1215	1					
11	2005年	1086	1					
12	2004年	995	0					
13	2003年	885	0					
14	2002年	819	0					
15								
16	统计养老金1000元的年份:							
17	输入公式: =COUNT(0/(\$B\$2:\$B\$14>=1000)), CTRL+shift+enter结束公式输入。							
18		10						

方法: 定位到某一单元格中, 输入公式: =COUNT(0/(\$B\$2:\$B\$14>=1000)), CTRL+shift+enter 结束公式输入。

答案: 10

此公式为数组公式, 与 C2:C14 中的公式相同, 但是是对 C2:C14 中的值进行加和汇总

不同的工作表计算

跨工作表计算

"!"

=AVERAGE(一班!B2,二班!B2,三班!B2,四班!B2)

跨工作簿计算

"[]"

=[工作簿间计算 A.xlsx]Sheet1!D2+[工作簿间计算 B.xlsx]Sheet1!D2

视频演示

跨工作表的计算

4 个班级的分数分别存在在 4 个工作表中

科目分数

语文 78

英语 90

数学 89

在汇总的单元格中输入=(> 切换到一班, 点对应班级的分数所在的单元格 > 切换到二班, 点对应班级的分数所在的单元格 > 三班, 四班 > 返回汇总的单元格 > 输入=(一班!B2+二班!B2+三班!B2+四班!B2)/4

跨工作簿的计算

工作簿间计算 A.xlsx
工作簿间计算 A.xlsx
工作簿间计算 A+B.xlsx

视频演示

Shift + Enter 同时打开多个工作簿

在三个工作簿都处于打开的状态下，在 A+B.xlsx 中定位到结果的单元格中 > 输入=> 切换到 A> 点击对应的语文成绩 > +> 切换到 B> 点击对应的英语成绩 > Enter

=[工作簿间计算 A.xlsx]Sheet1!\$D\$2+[工作簿间计算 B.xlsx]Sheet1!\$D\$2

> 双击填充柄 > 下面所有的数据都会变成与第一个同样的数据，因为所有的引用都是绝对引用 > F4 把引用转换为混合引用,对列进行绝对引用,对行进行相对引用 > =[工作簿间计算 A.xlsx]Sheet1!\$D2+[工作簿间计算 B.xlsx]Sheet1!\$D2 > 双击填充柄

每个人的成绩不一样，行变化而列不变，故只需要把列锁定即可。

公式追踪

追踪公式引用 单元格
追踪公式从属 单元格
追踪错误

视频演示

追踪公式引用

知识点：追踪公式引用单元格，即公式里数据的来龙去脉。

特点：不能保存，仅是实时的。

选中某个通过公式计算得到的单元格 > 公式 > 公式审核 > 追踪引用单元格，多次点击，直到找到最终的引用。

追踪引用单元格：本单元格中引用了哪些单元格

追踪从属单元格：有哪些单元格引用了本单元格中的数据

	A	B	C	D	E	F	G	H	I
1	2008年各项计算机产品销售表								
2		一月	二月	三月	四月	五月	六月	七月	总和
3	硬盘	30	30	12	35	23	14	85	229
4	内存	32	87	35	68	86	89	33	430
5	主板	54	7	68	75	42	67	46	359
6	光驱	69	65	25	45	69	35	38	346
7	刻录机	54	87	14	65	41	54	11	326
8	声卡	26	75	89	75	54	122	26	467
9	显示卡	32	123	65	25	32	98	58	433
10	调制解调器	54	21	25	14	54	24	46	238
11	CPU	36	65	36	16	36	66	35	290
12	键盘	95	123	95	33	95	81	61	583
13	鼠标	54	81	33	43	49	68	67	395
14	各月业绩	536	764	497	494	581	718	500	4096
15									
16	比例	13.09%	18.65%	12.13%	12.06%	14.18%	17.53%	12.35%	


追踪错误

方法：选中错误的单元格 > 公式审核 > 错误检查 > 追踪错误

找到原因是 I14 单元格是空值，完成 I14 单元格的计算，错误消失。

	A	B	C	D	E	F	G	H	I
1	2008年各项计算机产品销售表								
2		一月	二月	三月	四月	五月	六月	七月	总和
3	硬盘	30	30	12	35	23	14	85	229
4	内存	32	87	35	68	86	89	33	430
5	主板	54	7	68	75	42	67	46	359
6	光驱	69	65	25	45	69	35	38	346
7	刻录机	54	87	14	65	41	54	11	326
8	声卡	26	75	89	75	54	122	26	467
9	显示卡	32	123	65	25	32	98	58	433
10	调制解调器	54	21	25	14	54	24	46	238
11	CPU	36	65	36	16	36	66	35	290
12	键盘	95	123	95	33	95	81	61	583
13	鼠标	54	81	33	43	49	68	67	395
14	各月业绩	536	764	497	494	581	718	500	
15									
16	比例	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	#DIV/0!	

错误检查 > 错误检查 可以逐个的检查单元格中的错误。



圈释无效数据

数据验证的主要功能是：限制数据在输入时的格式和范围

数据 > 数据工具 > 数据验证 > 数据验证

数据验证一般都在数据输入或导入之前进行操作

在输入数据之前，选中需要进行数据验证的单元格区域 > 数据验证，设置选项卡，允许：整数，勾选忽略空值，数据：介于，最小值 0，最大值 10，> 此时在这些区域中，就不能输入设置的范围之外的数据了。

对于已经存在的数据，先选择单元格区域 > 设置数据验证 > 圈释无效数据，就可以把无效的数据显示出来

如果想要对单元格区域使用多个限制，就要使用自定义中的公式。

显示所有的公式与隐藏公式

公式 > 公式审核 > 显示公式

在显示公式的情况下，就可以把公式打印出来了。

有些情况下又希望公式被隐藏，如工资的计算方法 > 选中想要隐藏公式的所有单元格 > 数字 > 设置单元格格式 > 保护 > 取消勾选保护，勾选隐藏 > 审阅 > 更改 > 保护工作表 > 现在点击用公式计算出来的单元格，编辑栏就不显示出公式了，但此时其它的单元格也不能修改了。

逐步理解函数的实现

"公式求值"对话框

F9

显示公式

【公式 | 公式审核 | 显示公式 】

编辑公式

编辑栏

单元格双击

复制公式

填充柄

CTRL+C, CTRL+V

隐藏 + 保护工作表

对于复杂的嵌套函数,如果知道哪一步出错.

知识点:公式求值。

方法 1: 公式选项卡的"公式求值"

方法 2: 功能键 F9

公式求值是自动的执行, 先执行什么, 后执行什么, 而 F9 可以根据选择的公式执行计算.

如点击此公式所在的单元格

=LEFT(B2,1) & RIGHT("00" & RIGHT(B2,LEN(B2)-1),3) > 公式 > 公式审核 > 公式求值 > 点击求值, 就可以看到公式的执行步骤, 也可以看到函数出错的位置和原因.

LEN(B2), 得出 B2 单元格中文本的长度

RIGHT(B2,LEN(B2)-1), 从 B2 单元格中从右边开始取出来从第 2 位到最后一位的字符

"00" & RIGHT(B2,LEN(B2)-1), 把 00 与 B2 单元格中取出来的从第 2 位到最后一位的字符组合起来.

RIGHT("00" & RIGHT(B2,LEN(B2)-1),3), 取出来组合出来字符的最后 3 位.

LEFT(B2,1) & RIGHT("00" & RIGHT(B2,LEN(B2)-1),3), 把组合出来的字符的最后 3 位与原字符的第 1 位组合起来.

	A	B	C
1			
2		A7	A007
3		A29	A029
4		A43	A043
5		A122	A122
6		A317	A317
7		B3	B003
8		B20	B020
9		C5	C005
10		C33	C033
11		C144	C144

F9 功能

定位到含有公式的单元格中 > 双击在单元格中显示公式 > 选中一个完整的包含函数名称和参数的公式 > F9 > 就可以把选中的公式计算出来了
可以通过这种方法对函数的功能和使用方法进行理解.

=REPT(0,10-LEN(B2))&B2

以 0 来把原有字符串补足 10 位

="NO."&REPT(" ",10-LEN(B2))&B2

以空格把原有字符串补足 10 位，并在前面加上 NO.，达到 NO.左对齐，数字右对齐的效果

	A	B	C	D
1			=REPT(0, 10-LEN(B2))&B2	="NO."&REPT(" ", 10-LEN(B2))&B2
2		1146	0000001146	NO. 1146
3		435210	0000435210	NO. 435210
4		977947	0000977947	NO. 977947
5		112028	0000112028	NO. 112028
6		71003	0000071003	NO. 71003
7		959784	0000959784	NO. 959784
8		630336	0000630336	NO. 630336
9		53614	0000053614	NO. 53614
10		221071	0000221071	NO. 221071
11		86	0000000086	NO. 86
12				

公式分类

普通公式

数组公式

命令公式(定义为名称的)

课后作业 1-类型转换案例

知识点:文本型转换为数值型数字。

使用 6 个公式之一来转换:(同样适用于日期)

- a) = A1*1
- b) = A1/1
- c) = A1+0
- d) = A1-0
- e) = - -A1 减负运算 (第 1 个-是减法,第 2 个-是负数)
- f) = VALUE(A1)

逻辑型数字转换为数值型的 6 个公式:(当 A1 是逻辑值时)

- 1) = A1*1
- 2) = A1/1
- 3) = A1+0
- 4) = A1-0
- 5) = - -A1 减负
- 6) = N(A1) (N 函数)

课后作业 2-数据整理实例 1

行政部门绩效考核

郑冉冉工作能力 88.5 协调性 90.9 责任感 85.2 积极性 87.4

把标题提取出来作为标题

方法:使用分列来实现,现在 F 列实现,删除 e 列,再在 e 列对 D 列使用,以此类推。

函数

函数功能

介绍常用函数, 参及其设置的方法

统计函数

文本函数

日期和时间函数

IF , VLOOKUP, SUMIF

IF+AND+OR

VLOOKUP+多查找条件(&), 第 3 个参数 column(), 第 4 个参数 true 等

SUMIF

案例 (获取身份证信息)

函数

函数实际上是 Excel 预定义的一种内置公式. 它通过使用一些称为参数的特定数值来按 特定的顺序或结构执行计算. 函数可用于执行简单或复杂的计算

例如, ROUND 函数可将数字四舍五入, 在单元格中输公式" =ROUND(5.84,1)" 后按 Enter,

可得到四舍五入后的结果 5.8

函数结构: 函数的结构以函数名称开始, 后接左括号, 然后以逗号分隔输入参数, 最后是右括号.

函数名称: 如果要查看可用的函数列表, 单击一个单元格并按 Shift+F3

函数参数:

数字

文本

逻辑值(例如, TRUE 或 FALSE)

数组

错误值(例如, #N/A)

单元格引用(例如, A3, H3 : M3)

其他函数

函数难点

1. 选择函数, 如何选择函数, 积累, 学习一个积累一个, 掌握它的特点.
2. 参数特点, 有无参数, 参数多少, 参数特点.
3. 参数设置. 最难的是引用. 绝对引用与相对引用.

技巧, 输入包括参数在内的函数之后, 点击函数格式中的参数, 就可以高亮显示输入的函数中对应的参数.

函数的参数, 一是看函数的帮助信息, 二是在输入函数时看它的格式.

不要怕出错, 要有查找和排除错误的能力.

如果学习函数--思路

理解数据与表格

清楚数据之间的逻辑关系

表格结构决定思路, 思路决定函数, 函数决定结果

复杂的公式

核心就是函数的嵌套, 函数里面套函数,

函数的嵌套最多可以 64 层 (Excel2003-2007 为最多套 7 层).

函数输入方法

方法 1: 直接输入, 专业级的

然后按 Enter 或单击"√"

方法 2: 单击 fx 开始, 初学者

然后按 Enter 或单击"√"

定位到任一单元格中 > 开始 > 点击编辑栏中的 fx > 插入函数

函数工具与技巧

公式复制

拖曳填充柄

双击填充柄(向下到邻列第 1 个空单元格上方)

选择性粘贴公式

函数工具提示 (选项设置)

判断参数是否可以省略(带方括号的)

善用函数的帮助，看说明知道函数的作用，看语法知道函数的用法，看实例跟着学习。

Excel 常用函数

数学函数

INT()	取整	3.345	=INT(C2)	3
		-3.4	=INT(C3)	-4
MOD()	求余数	10	=MOD(C4,3)	1
		100	=MOD(C5,3)	1
ROUND()	四舍五入	21.3456	=ROUND(C6,3)	21.346
ABS()	取绝对值	-23.6	=ABS(C7)	23.6
SQRT()	算术平方根	4	=SQRT(C8)	2
RAND()	产生随机数		=RAND()	0.633695092
RANDBETWEEN()			=RANDBETWEEN(1,10)	2

INT

将数字向下舍入到最接近的整数

语法: INT(number)

Number 为需要进行向下舍入取整的实数

示例:

A	B	C	
1	数据	公式	结果
2	-9.05	=INT(A2)	-10

3	3.02	=INT(A3)	3
4	28.9	=INT(A4)	28

MOD

返回两数相除的余。结果正负号与同

语法: MOD(number,divisor)

Number 为被除数. Divisor 为除数.

示例:

	A	B
1	公式	结果
2	=MOD(3,2)	1
3	=MOD(-3,2)	1
4	=MOD(3,-2)	-1
5	=MOD(-3,-2)	-1

RAND

返回大于 等 0 及小于 1 的 均匀分布随机数,每次计算工作表时都将返回一个新数值

语法: RAND()

若要生成 a 与 b 之间的随机实数, 请使用: RAND()*(b-a)+a

如果要使用函数 RAND 生成一随机数,并且使之不单元格计算而改变可以在编辑栏中输入"=RAND()",保持编辑状态然后按 F9 , 将公式永久性地改为随机数

示例:

	A	B
1	公式	结果
2	=RAND()	
3	=RAND()*100	

ROUND

返回某个数字按指定位取整后的。

语法: ROUND(number,num_digits)

Number 需要进行四舍五入的数字。

Num_digits 指定的位数,按此进行四舍五入。

示例

	A	B	C
1	数据	公式	结果
2	2.15	=ROUND(A2,1)	2.2
3	2.15	=ROUND(A3,0)	2
4	-2.15	=ROUND(A4,1)	-2.2
5	-2.15	=ROUND(A5,-1)	0

SUM

返回某一单元格区域中所有数字之和。

语法: SUM(number1,number2, ...)

Number1, number2, ... 为 1 到 255 个要求和的参数。

自动求和 ALT+"+"

示例:

	A	B	C
1	数据	公式	结果
2	1	=SUM(A2:A3)	3
3	2	=SUM(A2,A5)	1
4	3	=SUM(1,2,3)	6
5	a	=SUM(4,TRUE)	5

SUMIF

根据指定条件对若干单元格求和。

语法: SUMIF(range,criteria,sum_range) Range

Range 为用于条件判断的单元格区域。

Criteria 为确定哪些单元格将被相加求和的条件,其形式可以数字, 表达式或 文本。例如,条件可以表示为 32, ">32" 或 "apples"。 Sum_range 是要求和的实际单元格。

示例:

	A	B	C	D
1	项目	支出	公式	结果
2	服装	238	=SUMIF(A2:A7,"服装",B2:B7)	352
3	医疗	110	=SUMIF(B2:B7,">100")	554
4	食品	58		
5	服装	28		

6	服装	86		
7	医疗	106		

统计函数

MAX()	求最大
MIN()	求最大
SUM()	求和
COUNT()	数值计数
COUNTA()	计数
AVERAGE()	求平均
COUNTIF()	条件计数
SUMIF()	条件求和
AVERAGEIF()	条件平均
COUNTIFS()	多条件计数
SUMIFS()	多条件求和
AVERAGEIFS()	多条件平均
FREQUENCY()	求数据分布频率
RANK()	排名次

AVERAGE

返回参数的平均值(算术)。

语法: AVERAGE(number1,number2, ...)

Number1, number2, ... 为需要计算平均值的 1 到 255 个参数。

示例:

	A	B	C
1	数据	公式	结果
2	1	=AVERAGE(A2:A3)	1.5
3	2	=AVERAGE(A2:A4)	1.5
4	(空单元格)	=AVERAGE(A2:A5)	1.5
5	a	=AVERAGE(1,3,3,4)	2.75

COUNT

COUNT

返回包含数字以及参列表中的单元格个。利用函 COUNT 可以计算单元格区域或数字组中数字字段的输入项个数。

语法: COUNT(value1,value2,...)

Value1, value2, ... 为包含或引用各种类型数据的参(1 到 255 个),但只有数字类型的数据才被计算。

示例:

	A	B	C
1	数据	公式	结果
2	销售	=COUNT(A2:A8)	3
3	39790	=COUNT(A5:A8)	2
4		=COUNT(A2:A8,2)	4
5	19		
6	22.24		
7	TRUE		
8	#DIV/0!		

COUNTIF

计算区域中满足给定条件的单元格个数。

语法: COUNTIF(range,criteria)

COUNTIF(要查找的区域, 要查找的内容)

Range 为需要计算其中满足条件的单元格数目区域。 Criteria 为确定哪些单元格将被计算在内的条件,其形式可以数字, 表达式或 文本。

示例:

种类	数量	公式	结果
苹果	32	=COUNTIF(A2:A5,"苹果")	2
柑桔	54	=COUNTIF(B2:B5,">50")	3
桃子	75		
苹果	86		

COUNTIF 函数的 18 种用法

1, 返回包含值 12 的单元格数量

=COUNTIF(A:A,12)

2, 返回包含负值的单元格数量

=COUNTIF(A:A,"<0")

3, 返回不等于 0 的单元格数量

=COUNTIF(A:A,"<>0")

4, 返回大于 5 的单元格数量

=COUNTIF(A:A,">5")

5, 返回等于单元格 B1 中内容的单元格数量

=COUNTIF(A:A,B1)

6, 返回大于单元格 B1 中内容的单元格数量

=COUNTIF(A:A,">"&B1)

7, 返回包含文本内容的单元格数量

=COUNTIF(A:A,"*")

8, 返回包含三个字符内容的单元格数量

=COUNTIF(A:A,"???")

9, 返回内容为单词"GOOD"(不区分大小写)的单元格数量

=COUNTIF(A:A,"GOOD")

10, 返回在文本中任何位置包含单词"GOOD"(不区分大小写)的单元格数量

=COUNTIF(A:A,"*GOOD*")

11, 返回内容以单词"AB"(不区分大小写)开头的单元格数量

=COUNTIF(A:A,"AB*")

12, 返回包含当前日期的单元格数量

=COUNTIF(A:A,TODAY())

13, 返回大于平均值的单元格数量

=COUNTIF(A:A,">"&AVERAGE(A:A))

14, 返回不等于空值和错误的单元格数量

=COUNTIF(A:A,"<>")

15, 返回包含值为 3 或-3 的单元格数量

=SUM(COUNTIF(A:A,{3,-3}))

16, 返回包含逻辑值为 TRUE 的单元格数量

=COUNTIF(A:A,TRUE)

17. 返回相同身份证号码的数量

=COUNTIF(A:A,A1&"*")

18. 返回区域内不重复的数量

=SUMPRODUCT(1/COUNTIF(A1:A20,A1:A20))

COUNTIF 统计迟到和旷课次数。

日期 学号	1月1日	1月2日	1月3日	1月4日	1月5日	迟到次数	旷课次数
2009102101	b	a	b	a	a	2	0
2009102102	a	a	a	a	a	0	0
2009102103	b	b	b	c	a	3	1
2009102104	a	c	a	b	a	1	1
2009102105	c	b	a	a	c	1	2
2009102106	a	a	a	a	a	0	0
2009102107	a	c	a	a	a	0	1
2009102108	a	a	c	a	a	0	1
2009102109	a	a	c	a	a	0	1
2009102110	a	a	a	a	a	0	0

a:全勤

b:迟到

c:旷课

迟到次数 =COUNTIF(B2:F2,"b")

旷课次数 =COUNTIF(B2:F2,"c")

COUNTIF 统计 80-89 之间学生的数目

姓名	数据库技术与编程
谭高鑫	82.8
宋志阳	59.2
王千慧	68
高艺	92
蔡仲璋	85.8
李文琪	99
庞振军	75.6
黄舒馨	89
黄武聪	80.6
陈滢歆	80.8

刘志来	80
范凯媛	73

使用 COUNTIF 函数

=COUNTIF(B2:B13,">="&D2)-COUNTIF(B2:B13,">"&E2)

注意: 因为 COUNTIF 只有一个参数,所以要用相减的方法来实现, 由于是统计这个区间中的人数, 要包含 89, 所以在减号后面的条件中不能写成是>, 否则就把 89 分的排除在外了.

使用 COUNTIFS 函数

=COUNTIFS(P2:P122,">=80",P2:P122,"<89")

或

=COUNTIFS(P2:P122,">="&80,P2:P122,"<=89")

注意: 在 COUNTIFS 函数中, 后面的条件则要写为<=才能把 89 分的包括在内.

可以把 80 和 89 改为单元格引用, 这样就方便进行其它值的统计, 还可以把范围做成一个列表, 就可以使用填充柄拖动来计算一系列范围中值的统计, 注意使用文本的连接符号&

统计区间		人数 (COUNTIF)	人数 (COUNTIFS)
70	89	8	8

=COUNTIF(B2:B13,">="&D2)-COUNTIF(B2:B13,">"&E2)

=COUNTIFS(B2:B13,">="&D2,B2:B13,"<="&E2)

COUNTIF 多条件求和

学号	姓名	性别	数学	计算机	英语
201210201	李孔	男	78	100	88
201210208	马克	男	62	92	89
201210203	张扬	男	93	91	72
201210204	严冬	男	59	78	83
201210209	董齐	男	78	56	90
201210202	李文	女	90	90	90
201210207	丘陵	女	100	87	87
201210205	刘柳	女	81	59	75

201210206	夏令	女	99	92	67
-----------	----	---	----	----	----

1.数学与计算机考试都大于 90 的人数:

=COUNTIFS(E2:E10,">90",F2:F10,">90")

2.数学与计算机考试都大于 90 的男生人数:

=COUNTIFS(E2:E10,">90",F2:F10,">90",D2:D10,"男")

日期 学号	3 月 6 日	3 月 7 日	3 月 8 日	3 月 9 日	3 月 10 日	迟到次数	旷课次数	正常上课
20120001	☆	●	☆	☆	☆	0	1	4
20120002	☆	☆	▲	☆	●	1	1	3
20120003	☆	☆	☆	☆	☆	0	0	5
20120004	☆	☆	☆	☆	☆	0	0	5
20120005	☆	☆	▲	☆	☆	1	0	4
20120006	▲	☆	☆	●	☆	1	1	3
20120007	☆	☆	☆	☆	☆	0	0	5
20120008	☆	☆	●	☆	▲	1	1	3
20120009	☆	☆	☆	☆	☆	0	0	5

注释:▲三角形:迟到; ●圆:旷课。

=COUNTIFS(B2:F2,"▲")

=COUNTIFS(B2:F2,"●")

=COUNTIFS(B2:F2,"☆")

SUMIF

让 SUMIF 函数忽略求和区域中的错误值

=SUMIF(A2:A10,"<=9E+307")

求达到 90 的数据和

=SUMIF(A2:A12,">=90")

替 SUMIF 函数设置可变的求和条件

=SUMIF(\$G\$2:\$I10,">="&I4)

在 SUMIFS 函数中使用通配符

=SUMIFS(C2:C11,A2:A11,"张三",B2:B11,"*电视机*")

让 SUMIF 函数按模糊条件对数据求和

=SUMIF(F2:F10,I2&"*",G2:G10)

4 个字符:
=SUMIF(F2:F10,"????",G2:G10)

MAX

MAX
返回一组值中的最大。
语法: MAX(number1,number2, ...)
Number1, number2, ... 是要从中找出最大值的 1 到 255 个数字参数。

示例:

	A	B	C
1	数据	公式	结果
2	1	=MAX(A2:A3)	2
3	2	=MAX(A2:A4)	2
4	(空单元格)	=MAX(A2:A5)	2
5	a	=MAX(1,3,3,4)	4

文本函数

MID()	取子串	国庆六十周年	=MID(C1,3,4)	六十周年
LEFT()	从左取子串	国庆六十周年	=LEFT(C2,3)	国庆六
RIGHT()	从右取子串			
LEN()	文本长度	国庆六十周年	=LEN(C4)	6
TEXT()	数字转化文本格式	90	=TEXT(C5,"0")	90
REPT	文本重复		=REPT("01",3)	010101
REPLACE	替换特定位置的文本	国庆六十周年	=REPLACE(C7,3,2,"60")	国庆 60 周年
SUBSTITUTE	替换文本	国庆六十周年	=SUBSTITUTE(C8,"六十","60")	国庆 60 周年
FIND	查找文本所在位置	国庆六十周年	=FIND("六十",C9,2)	3

LEN

LEN
返回文本字符串中的数

语法: LEN(text)
text 是要查找其长度的文本。
空格将作为字符进行计数

示例:

	A	B	C
1	数据	公式	结果
2	Excel	=LEN(A2)	5
3	(3 个空格)	=LEN(A3)	3
4	清单	=LEN(A4)	2

LEFT

LEFT
基于 所指定的字符
数返回文本串中第一个或前几语法：
LEFT(text,num_chars)
Text 是包含要提取字符的文本串。
Num_chars 指定要由 LEFT 所提取的字符数。

示例:

	A	B	C
1	数据	公式	结果
2	北京市 海淀区	=LEFT(A2,2)	北京
3		=LEFT(A2)	北

MID

MID
返回文本字符串中从指定位置开始的特定数目的字符串
语法: MID(text,start_num,num_chars)
Text 是包含要提取字符的文本串。
start_num 是文本中要提取的第一个字符位置,
start_num 为 1,以此类推
Num_chars 指定希望 MID 从文本中返回字符的个数。

示例:

	A	B	C
--	---	---	---

1	数据	公式	结果
2	北京市 海淀区	=MID(A2,3,3)	市 海
3		=MID(A2,5,10)	海淀区
4		=MID(A2,10,5)	

TEXT 函数用法

特色

相当于整容函数

返回的是文本类型

格式

=TEXT(A2,"正数;负数;零;文本 ") --- 四个区段

=TEXT(A2,"非负数;负数") --- 二个 区段

=TEXT(A2,"正数;负数;零") --- 三个区段

=TEXT(A2,"条件 1;条件 2;不满足条件 1,2;文本") --- 四个区段

如=TEXT(B2,"[>=60]及格;[>=90]优秀;不及格;分数错误")

=TEXT(A2,"000") --- 固定位数, 不足则被 0

=TEXT(A2,"????.???") --- 小数点对齐

=TEXT(A2,"####.####") --- 去掉数字中无意义的 0, 只保留有效位数

=TEXT(A2,"0,") -- =缩小 1000 倍; =TEXT(A2,"0,") -- =缩小 10000000 倍

=TEXT(A2,"[DBNum2]G/通用格式") --- 转为中文大写

FIND, FINDB 函数

本文介绍 Microsoft Excel 中 FIND 和 FINDB 函数的公式语法和用法。

说明

函数 FIND 和 FINDB 用于在第二个文本串中定位第一个文本串，并返回第一个文本串的起始位置的值，该值从第二个文本串的第一个字符算起。

重要:

这些函数可能并不适用于所有语言。

FIND 适用于使用单字节字符集 (SBCS) 的语言，而 FINDB 适用于使用双字节字符集 (DBCS) 的语言。 您的计算机上的默认语言设置对返回值的影响方式如下:

无论默认语言设置如何,函数 FIND 始终将每个字符(不管是单字节还是双字节)按 1 计数。

当启用支持 DBCS 的语言的编辑并将其设置为默认语言时,FINDB 会将每个双字节字符按 2 计数。 否则,FINDB 会将每个字符按 1 计数。

支持 DBCS 的语言包括日语,中文(简体),中文(繁体)以及朝鲜语。

语法

FIND(find_text, within_text, [start_num])

FINDB(find_text, within_text, [start_num])

FIND 和 FINDB 函数语法具有下列参数:

find_text 必需。 要查找的文本。

within_text 必需。 包含要查找文本的文本。

start_num 可选。 指定开始进行查找的字符。 within_text 中的首字符是编号为 1 的字符。 如果省略 start_num,则假定其值为 1。

备注

FIND 和 FINDB 区分大小写,并且不允许使用通配符。 如果您不希望执行区分大小写的搜索或使用通配符,则可以使用 SEARCH 和 SEARCHB 函数。

如果 find_text 为空文本(""),则 FIND 会匹配搜索字符串中的首字符(即编号为 start_num 或 1 的字符)。

Find_text 不能包含任何通配符。

如果 within_text 中没有 find_text,则 FIND 和 FINDB 返回 错误值 #VALUE!。

如果 start_num 不大于 0,则 FIND 和 FINDB 返回 错误值 #VALUE!。

如果 start_num 大于 within_text 的长度,则 FIND 和 FINDB 返回 错误值 #VALUE!。

可以使用 start_num 来跳过指定数目的字符。 以 FIND 为例,假设要处理文本字符串 "AYF0093.YoungMensApparel"。 若要在文本字符串的说明部分中查找第一个"Y"的编号,请将 start_num 设置为 8,这样就不会搜索文本的序列号部分。 FIND 从第 8 个字符开始查找,在下一个字符处找到 find_text,然后返回其编号 9。 FIND 始终返回从 within_text 的起始位置计算的字符编号,如果 start_num 大于 1,则会对跳过的字符计数。

示例

复制下表中的示例数据，然后将其粘贴进新的 Excel 工作表的 A1 单元格中。要使公式显示结果，请选中它们，按 F2，然后按 Enter。如果需要，可调整列宽以查看所有数据。

数据		
Miriam McGovern		
公式	说明	结果
=FIND("M",A2)	单元格 A2 中第一个"M"的位置	1
=FIND("m",A2)	单元格 A2 中第一个"M"的位置	6
=FIND("M",A2,3)	从单元格 A2 的第三个字符开始查找第一个"M"的位置	8

示例 2

数据		
Ceramic Insulators #124-TD45-87		
Copper Coils #12-671-6772		
Variable Resistors #116010		
公式	描述(结果)	结果
=MID(A2,1,FIND("#",A2,1)-1)	提取单元格 A2 中从第一个字符到"#"的文本 (Ceramic Insulators)	Ceramic Insulators
=MID(A3,1,FIND("#",A3,1)-1)	提取单元格 A3 中从第一个字符到"#"的文本 (Copper Coils)	Copper Coils
=MID(A4,1,FIND("#",A4,1)-1)	提取单元格 A4 中从第一个字符到"#"的文本 (Variable Resistors)	Variable Resistors

Replace 函数隐藏几位手机号

手机号	隐藏 8-10 位
18996471864	1899647***4
18996481864	1899648***4
18996491863	1899649***3
18996502020	1899650***0
18996512020	1899651***0
18996522020	1899652***0
18996532020	1899653***0
18996542019	1899654***9
18996552019	1899655***9

方法:

=REPLACE(A2,8,3,"***"), A2 为需要进行替换的原文本所在所在单元格, 8 为从第几位开始

替换, 3 为替换几位, ***表示替换为的文本.

=REPLACE(A2,8,3,REPT(" ",3))

REPT 函数, 将文本重复多次

逻辑函数

逻辑函数:	
IF()	判断
AND()	与
OR()	或
NOT()	非

IF

执行真假值判断, 根据逻辑计算的返回不同结果。

语法: (IF(logical_test,value_if_true,value_if_false))

Logical_test 表示计算结果为 TRUE 或 FALSE 的任意值或表达式

Value_if_true logical_test 为 TRUE 时返回的值。

Value_if_false logical_test 为 FALSE 时返回的值。

示例:

	A	B	C
1	数据	公式	结果
2	50	=IF(A2<60,"不及格","及格")	不及格
3	70	=IF(A3>=85,"A",IF(A3>=75,"B",IF(A3>=60,"C","D")))	C
4	90	=IF(A4>=85,"A",IF(A4>=75,"B",IF(A4>=60,"C","D")))	A

视频演示

IF+AND+OR 例子

班级	学号	姓名	性别	数学	计算机	英语	评价(三门均通过)	评价(三门之一通过)
902	90201	李孔	男	78	100	88	通过	通过
902	90202	孙文	女	90	90	90	通过	通过
902	90203	张杨	男	90	89	72	通过	通过
903	90301	严冬	男	56	78	83	不通过	通过

903	90302	刘柳	女	78	76	75	通过	通过
903	90303	夏令	女	99	75	67	通过	通过
904	90401	丘陵	女	100	87	87	通过	通过
904	90402	马克	男	62	92	89	通过	通过
904	90403	董齐	男	78	56	58	不通过	通过

=IF(AND(E2>=60,F2>=60,G2>=60),"通过","不通过")

=IF(OR(E2>=60, F2>=60, G2>=60), "通过", "不通过")

对单科成绩进行评价, 60 分以下为不及格, 60-74 分为及格, 75-84 分为良好, 85-100 分为优秀.

成绩	评价	IF()	VLOOKUP()
59	不及格	不及格	不及格
60	一般	一般	一般
84	良好	良好	良好
100	优秀	优秀	优秀

=IF(A2<60,"不及格",IF(A2<75,"一般",IF(A2<85,"良好","优秀")))

从小到大判断语法会比较简单, 先判断是否小于 60, 如果小于 60, 不及格, 就不会再进行下面的判断. 如果大于 60, 再判断是否小于 75, 如果小于 75, 就是及格, 如此依次进行判断.

=VLOOKUP(A2,\$A\$2:\$B\$5,2)

Excel 不支持写为 IF(60<=M6<=70)这种写法

日期函数

YEAR()	求年		
MONTH()	求月		
DAY()	求日		
TODAY()	当前日期	=TODAY()	2018/3/27
DATE()	计算给定的日期	=DATE(2011,11,21)	2011/11/21
NOW()	当前系统的日期和时间	=NOW()	2018/3/27 15:31
DATEDIF()	计算日期差		
NOW()-TODAY()	计算当前是几点几分	=NOW()-TODAY()	15:31:29
MOD(NOW(),1)	计算当前是几点几分	=MOD(NOW(),1)	15:31:29

YEAR(TODAY())	取当前日期的年份	=YEAR(TODAY())	2018
MONTH(TODAY())	取当前日期的月份	=MONTH(TODAY())	3
DAY(TODAY())	计算当前日期是几号	=DAY(TODAY())	27
WEEKDAY(TODAY(),2)	计算今天是星期几	=WEEKDAY(TODAY(),2)	2
第二参数 2, 表示将星期一计算为 1, 这样比较符合国人的习惯. 如果省略第二个参数, 则表示将星期日计算为 1, 星期六计算为 7			
TEXT(TODAY(),"aaa")	计算今天是星期几	=TEXT(TODAY(),"aaa")	二
EDATE()	指定日期前后月份的日期		
EOMONTH	某个月份最后一天的序列号		

YEAR

65

YEAR

返回某日期对应的年份。值为 1900 到 9999 之间的整数

语法: YEAR(serial_number)

Serial_number 为一个日期值,其中包含要查找年份的。应使用 DATE DATE DATE DATE 函数来输入日期,或者将日作为其他公式函数的结果输入。例如使用 DATE(2008,5,23)输入 2008 年 5 月 23 日。

示例:

	A	B
1	公式	结果
2	=YEAR(DATE(2008,8,8))	2008
3	=YEAR("2005-7-4")	2005

8, =EDATE(TODAY(),1)

计算当前日期之后一个月的。

如果第二参数写成 -1, 就是计算当前日期之 1 个月的日期。

9, =EOMONTH(TODAY(),1)

计算下个月最后一天的日期。

如果第二参数写成 0, 是本月最后一天的日期。用 =DAY(EOMONTH(TODAY(),0))计算出当前这个月一共有多少天了。

10, =EOMONTH(TODAY(),0)-TODAY()

计算今天到本月底还有多少。

使用 DATEDIF 计算日期差值

起始日期	结束日期	差	公式	参数	说明
2010/8/1	2012/2/4	1	=DATEDIF(A2,B2,"Y")	Y	相差年数
2010/8/1	2012/2/4	18	=DATEDIF(A3,B3,"M")	M	相差总月数
2010/8/1	2012/2/4	6	=DATEDIF(A4,B4,"YM")	YM	一年内相差月数
2010/8/1	2012/2/4	552	=DATEDIF(A5,B5,"D")	D	相差总天数
2010/8/1	2012/2/4	187	=DATEDIF(A6,B6,"YD")	YD	一年内相差天数
2010/8/1	2012/2/4	3	=DATEDIF(A7,B7,"MD")	MD	一月内相差天数
		0	=DATEDIF("2012-6-30","2012-6-1","y")		

使用函数 WEEKDAY 来判断是否为周末

日期	是否周末
2017/11/18	TRUE
2017/11/23	FALSE
2017/11/24	FALSE
2017/11/25	TRUE
2017/12/3	TRUE
2017/12/4	FALSE
2017/12/5	FALSE
2017/12/6	FALSE
2017/12/7	FALSE

=WEEKDAY(A2,2)>5

WEEKDAY 的第二参数 2, 表示将星期一计算为 1, 这样比较符合国人的习惯. 如果省略第二个参数, 则表示将星期日计算为 1, 星期六计算为 7

6,7 为周末, 如果是周末则返回 TRUE, 周一到周五返回 FALSE.

WORKDAY()计算某日期第 n 个工作日之后的日期

日期	第 n 个工作日	第 n 个工作日的日期
2017/9/30	10	2017/10/20
节假日期	原因	
2017/10/1	国庆节	
2017/10/2	国庆节	
2017/10/3	国庆节	

2017/10/4	国庆节	
2017/10/5	国庆节	
2017/10/6	国庆节	
2017/10/7	国庆节	

2017/9/30 日, 计算其 10 个工作日之后的日期, 注意要跳过 10 月 1 日到 10 月 7 日国庆节放假的日期.

WORKDAY()

返回在某日期(起始日期)之前或之后, 与该日期相隔指定工作日的某一日期的日期值. 工作日不包括周末和专门指定的假日.

=WORKDAY(A2, B2, A5:A11)

A2 为起始日期 2017/9/30, B2 为间隔的日期值 10, A5:A11 为指定的 7 天国庆假期.

METWORKDAYS

开始日期和结束日期之间有多少个工作日

开始日期	结束日期	有多少个工作日
2017/9/30	2017/10/20	10
节假日	原因	
2017/10/1	国庆节	
2017/10/2	国庆节	
2017/10/3	国庆节	
2017/10/4	国庆节	
2017/10/5	国庆节	
2017/10/6	国庆节	
2017/10/7	国庆节	

=NETWORKDAYS(A2,B2,A5:A11)

A2 为起始日期 2017/9/30, B2 为结束日期 2017/10/20, A5:A11 为指定的节假日.

已知入职日期和试用期, 使用 EDATE()计算转正日期

姓名	入职日期	试用期(月)	转正日期
张三	2017/11/18	3	2018/2/18

=EDATE(B2, C2)

返回与指定日期相隔 n 个月的日期值.

判断上中下旬

日期	判断上中下旬
2017/11/18	中旬
2017/11/23	下旬
2017/11/24	下旬
2017/11/25	下旬
2017/12/3	上旬
2017/12/4	上旬
2017/12/5	上旬
2017/12/6	上旬
2017/12/7	上旬

=LOOKUP(DAY(A2),{0,11,21},{"上旬","中旬","下旬"})

计算某月最后 1 天，某月多少天，判断第几季度。

签订日期	该月最后一天	该月最后一天	该月有多少天	第几季度
2011/12/10				
2013/2/15				
2012/4/19				
2012/8/1				
2011/10/1				
2013/6/22				
2013/9/1				
2012/4/19				
2012/3/1				
2011/10/1				

该月最后一天，方法一

=DATE(YEAR(A2),MONTH(B3)+1,0)

用 YEAR(A2)选取单元格中的年，用 MONTH(A2)选取单元格中的月，MONTH(A2)+1 表示下一个月，而 0 表示当月的第 0 天，也就是上个月的最后一天，总体还表示单元格中日期所在月的最后一天。

该月最后一天, 方法二

=EOMONTH(A2,0)

EOMONTH, 用来求一个月的最后一天.

第一个参数表示开始日期, 第二个参数为 **months**, 如果等于 0, 表示当月, 如果等于 1, 表示下个月, 如果等于-1, 表示上个月的, 如此类推.

该月有多少天

=DAY(EOMONTH(A2,0))

DAY 函数返回某个日期离 1 日有多少天, 只要以该月最后一天为参数就能得到该月有多少天.

如果返回的是日期, 格式化为常规.

第几季度

季度是与月份相关的, 先取出日期中的月份, 除以 3, 再对结果的整数加 1, 就能得到季度了. 可以使用 ROUNDUP 函数向上舍入. 0 表示保留到整数位.

=ROUNDUP(MONTH(A2)/3,0)

再使用 TEXT 函数来拼接第几季度的文本

=TEXT(ROUNDUP(MONTH(B3)/3,0),"第 0 季度")

或

= "第"&TEXT(ROUNDUP(MONTH(B3)/3,0),"0 季度")

MOD 判断闰年

作业:IF+MOD 应用,统计闰年或平年。

提示:MOD\AND\OR

年份	是否闰年
2000	
2001	
2002	
2003	
2004	
2005	

2006	
2007	
2008	
2009	
2000	
2001	

如果能被 100 整除, 且能被 400 整除, 则为闰年

=AND(MOD(A2,100)=0,MOD(A2,400)=0)

如果能被 4 整除, 且不能被 100 整除, 则为闰年

=AND(MOD(A2,100)<>0,MOD(A2,4)=0)

OR(AND(MOD(A2,100)=0,MOD(A2,400)=0),AND(MOD(A2,100)<>0,MOD(A2,4)=0))

判断是否为闰年

=IF(OR(AND(MOD(A2,100)=0,MOD(A2,400)=0),AND(MOD(A2,100)<>0,MOD(A2,4)=0)),

闰年","平年")

注意:

=IF(COUNT("-2-29"),"是","否"), 为什么使用 F9 与使用公式求值得出的结果不一样?

在 Excel 中如果输入 "月/日" 形式的日期, 会默认按当前年份处理. 如果当前年中没有 2 月 29 日, 输入 "2-29" 就会作为文本处理. 如果当前年份没有 2 月 29 日, "2-29" 前面加上负号, 就相当于在文本前加负号, 会返回错误值 #VALUE!. 再用 COUNT 函数判断 "-2-29" 是数值还是错误值, 如果是错误值, 当然就不是闰年了. 注意, "2-29" 的写法在不同语言版本中结果可能有差异.

停车收费系统

	A	B	C	D	E	F	G	H	I	J
1	停车收费系统									
2	车牌号	户主	停车日期	停车时间	离开日期	离开时间	停留时间(天)	停留时间(小时分秒)	停留时间(分)	收费(元)
3	京PR6K37	褚一一	2015/2/1	21:09:09	2015/2/2	3:18:02				
4	京PR6K38	杨一二	2015/2/2	21:14:03	2015/2/3	13:08:02				
5	京PR6K20	蒋一三	2015/2/3	21:56:09	2015/2/5	23:08:02				
6	京PR6K40	沈一四	2015/2/1	18:09:45	2015/2/2	3:12:02				
7	南K12001	韩一五	2015/2/2	19:09:40	2015/2/3	3:08:02				
8	南K12002	许一六	2015/2/3	20:09:45	2015/2/4	13:08:02				
9	南K12803	朱一七	2015/2/4	21:09:09	2015/2/5	23:08:02				
10	南K12004	秦一八	2015/2/1	21:56:09	2015/2/2	6:18:02				
11	浙A1063	尤一九	2015/2/6	17:09:12	2015/2/10	8:08:02				
12	浙A1064	张二〇	2015/2/1	19:09:30	2015/2/2	13:08:02				
13	浙A1065	何二一	2015/2/8	21:56:09	2015/2/12	23:08:02				
14	浙A1066	吕二二	2015/2/3	18:09:31	2015/2/15	13:08:02				

停车收费系统									
车牌号	户主	停车日期	停车时间	离开日期	离开时间	停留时间 (天)	停留时间 (时分秒)	停留时间 (分)	收费 (元)
京 PR6K37	褚一一	2015/2/1	21:09:09	2015/2/2	3:18:02				
京 PR6K38	杨一二	2015/2/2	21:14:03	2015/2/3	13:08:02				
京 PR6K20	蒋一三	2015/2/3	21:56:09	2015/2/5	23:08:02				
京 PR6K40	沈一四	2015/2/1	18:09:45	2015/2/2	3:12:02				
南 K12001	韩一五	2015/2/2	19:09:40	2015/2/3	3:08:02				
南 K12002	许一六	2015/2/3	20:09:45	2015/2/4	13:08:02				
南 K12803	朱一七	2015/2/4	21:09:09	2015/2/5	23:08:02				
南 K12004	秦一八	2015/2/1	21:56:09	2015/2/2	6:18:02				
浙 A1063	尤一九	2015/2/6	17:09:12	2015/2/10	8:08:02				
浙 A1064	张二〇	2015/2/1	19:09:30	2015/2/2	13:08:02				
浙 A1065	何二一	2015/2/8	21:56:09	2015/2/12	23:08:02				
浙 A1066	吕二二	2015/2/3	18:09:31	2015/2/15	13:08:02				

知识点:时间计算方法。

作业:计算停车费用。

函数:IF, HOUR, MINUTE,SECOND

收费标准:

1. 每分钟 0.1 元;
2. 30 秒及以上算 1 分钟。

停留时间(天)

(离开日期+离开时间)-(停车日期+停车时间)

$G3 = \text{INT}((E3 + F3) - (C3 + D3))$

停留时间(小时分秒)

$H3 = (E3 + F3) - (C3 + D3)$ > 开始 > 数字 > 数字格式 > 时间

停留时间(分)

$I3 = G3 * 24 * 60 + \text{HOUR}(H3) * 60 + \text{MINUTE}(H3) + \text{IF}(\text{SECOND}(H3) \geq 30, 1, 0)$ > 开始 > 数字 >

减少小数位数

收费

$= I3 * 0.1$

生成间断序号

在 A 列中输入间断序号, 以 B 列为参考

姓名	性别	部门	签订日期	期限(年)	到期日	COUNTA 方法	MAX 方法
赵一	男	办公室	2011/12/10	4	2015/12/9	1	1
钱二	男	办公室	2013/4/15	2	2015/4/14	2	2
孙三	女	销售部	2012/4/19	3	2015/4/18	3	3
王四	男	销售部	2012/8/1	3	2015/7/31	4	4
周五	男	销售部	2011/10/1	4	2015/9/30	5	5
吴六	女	财务部	2013/6/22	2	2015/6/21	6	6
郑七	女	财务部	2013/9/1	2	2015/8/31	7	7
卫八	男	财务部	2012/4/19	5	2017/4/18	8	8
冯九	男	技术部	2012/8/1	5	2017/7/31	9	9
陈一	女	技术部	2011/10/1	4	2015/9/30	10	10
褚一一	男	技术部	2013/6/22	5	2018/6/21	11	11
杨一二	男	技术部	2013/9/1	5	2018/8/31	12	12
蒋一三	女	技术部	2012/4/19	5	2017/4/18	13	13
沈一四	女	技术部	2012/8/1	7	2019/7/31	14	14
韩一五	女	技术部	2011/10/1	5	2016/9/30	15	15

使用 COUNTA 函数

=IF(B2="", "", COUNTA(A\$2:A2))

COUNTA 函数计算范围中不为空的单元格的个数。

如果 A 列中对应的单元格是空值, 得到的序号列中也为空, 如果 A 列中不为空, 则序号的值为当前行对应的 A 列中的单元格与 A2 单元格之间非空单元格的个数。

使用 MAX 函数

=IF(A2="", "", MAX(G\$1:G1)+1)

如果 A 列中对应的单元格是空值, 序号列中也为空, 如果 A 列中不为空, 则序号列中的值为当前行与 G1 单元格之间单元格区域中的最大值+1, MAX 函数会跳过字符串和空值。对于第 2 行, A2 不为空, 就会返回 MAX(G\$1:G1)+1 中的值。但 G\$1:G1 中的内容无数字, 无法比较大小, 故 MAX(G\$1:G1)返回 0, MAX(G\$1:G1)+1 返回 1, 第二行中的序号为 1, 对于 G3, G\$1:G2 最大值为 1, 返回序号值为 2, 第 4 行为空, 则序号也返回为空, 等等。

也可以对 H 列使用, 这种方法不依赖于 G 列的数据, 更具有技巧性, 推荐使用这种方法。

=IF(A2="", "", MAX(H\$1:H1)+1)

引用与查找函数

VLOOKUP()	垂直方向查找
OFFSET()	计算偏移量
MATCH()	
INDEX()	
INDIRECT()	文本字符串指定的引用
ROW()	引用行的数据
COLUMN()	引用列的数据
HLOOKUP()	水平方向查找
Choose()	索引值指定参数列表中的数值。
Find()	一个字符串在另一个字符串的起始位置

OFFSET 引用函数(用于生成动态图表, 动态透视表等)

MATCH 引用函数

INDEX 查找函数

ROW 和 COLUMN 获取行, 列号

INDIRECT 函数

VLOOKUP, HLOOKUP 查找函数

SUMPRODUCT 数组函数

Choose() 索引值指定参数列表中的数值。(自学)

Find()一个字符串在另一个字符串的起始位置(自学)

OFFSET 函数, INDIRECT 函数, SUMPRODUCT 函数是中级水平验明正身的标杆函数, 不会这三个难说自己已经是中级水平。

CHOOSE 函数

业绩核算-CHOOSE

CHOOSE 函数语法:

排在第三个那个,给我站出来

=CHOOSE(B2,"TOM","MARRY","JONE","JACK","KATE")

姓名部门补助

TOM 交通部 1000
MARRY 安监部 1200
JONE 储运部 1500
JACK 卫生部 1600
KATE 卫生部 1601

3

JONE

=CHOOSE(E25,"TOM","MARRY","JONE","JACK","KATE")

FIND 函数

FIND 函数语法:

=FIND(找啥,在哪段字里找,从第几个字开始往后找)

员工产品销售金额表

姓名产品(单位:元)提成总额

路由器服务器台式电脑笔记本电脑液晶电脑

田洁 1320000378000122500201600150400

何宁 1430000270000126000223200206800

陈雨 88000025200015050013680089300

肖牧 121000028800087500273600164500

毕伟 99000030600010850018720098700

祝雪 770000342000126000165600178600

销售指标(元)1300000276000216000374000360000

提成方案

销售额区间应提成比例

01000010%

10001250008%

250011500006%

1500013000004%

30000110000002%

10000011%

业绩核算:

输入姓名何宁

输入产品名称台式电脑

销售额 126000

是否完成任务未完成

应提成额 7560

Offset()

功能: 通过给定偏移量得到新的引用区域。

语法: OFFSET(reference, rows, cols,[height], [width])

有 5 个参数:

Reference —— 基点或参照系, 即引用区域左上角单元格;

Rows —— 偏移的行数 ,可以无, 必须留位置

行数正数: 在基点的下方; 负数:在基点的上方

Cols —— 偏移的列数, 可以无, 必须留位置。

列数正: 基点的右边; 负数:在基点的左边。

Height—— 高度, 即返回的引用区域的行数, 必须为正数。

Width—— 宽度, 即所要返回的引用区域列数, 必须为正数

使用特点

设置所需要的区域

若结果为 1 个单元格, 其值显示在公式所在单元格

若结果为多行多列, 使用数组公式方法, 其值显示所在区域

使用在

单元格区域

求和, 平均最大小的统计区域

查找函数的范围参数

视频演示

OFFSET 函数


OFFSET(reference,rows,cols,[height],[width])

5 个参数

1. 作为基准的单元格
2. 向下移动 x 格,向下移动为正,向上移动为负
3. 向右移动 y 格,向右移动为正,向左移动为负
4. 返回单元格区域 n 行]
5. 返回单元格区域 z 列]

以指定的引用为参照系, 通过给定偏移量得到新的引用。返回的引用可以为一个单元格或单元格区域。并可以指定返回的行数或列数。

Offset 理解图示 1



注意事项

- 1, reference 必须为引用(包括函数产生的三维引用),不能为常量或数组。
- 2, 行偏移_rows, 列偏移_cols 为必须项, 如省略必须用 "," 逗号进行占位, 缺省值为 0(即不偏移); 行数_height 和列数_width 为可选项, 可省略, 缺省视为与基点相同。

一, 基点与行数, 列数的关系。

通常情况下, 如果基点以单元格区域作为引用范围, 如果不指定行数或列数, 系统默认是引用区域的行数和列数来确定结果。

使用公式求值来得到结果引用的单元格, 使用 F9 只能得到数据

公式 1 =OFFSET(A1:C1,2,2) 公式结果 \$C\$3:\$E\$3

公式 2 =OFFSET(A1,2,2,,3) 公式结果 \$C\$3:\$E\$3

由于引用基点为 A1:C1 3 列区域, 同时公式 1 中未指定行数或列数, 所以最终结果仍然为 3 列区域, 公式 2 中没有指定取的行数, 与基点的相同, 即为 1 行, 取 3 列, 结果与公式 1 相同, 两者等价。

公式 3=OFFSET(A1:C5,2,2) 公式结果 \$C\$3:\$E\$7

公式 4=OFFSET(A1,2,2,5,3) 公式结果\$C\$3:\$E\$7

同理, 公式 3 与公式 4 结果相同, 两个公式等价。

二, 行数与列数对引用基点的影响。

在上面的公式 1 和公式 3 中, 都是基点为单元格区域引用, 同时都未设置行数与列数的情况下的结果。

但有时, 基点为区域引用时, 用户同时又指定了行数或(和)列数, 那么用户必须清楚了解实际引用的基点区域, 才能得出正确结果。

序号	姓名	工资
1	A	100
2	B	200

3	C	300
4	D	400
5	E	500

例如已经定义名称 DATA:=\$A\$2:\$C\$6

例 1: 利用名称作为基点, 得到右图的浅蓝色子区域公式:

=OFFSET(DATA,1,,2) 公式结果\$A\$3:\$C\$4, 检验:1

由于公式中指定了最终结果的行数(2), 未指定列数, 那么公式实际上以 A2:C2 区域作为引用基点, 公式等同于:

公式 1 =OFFSET(A2:C2,1,,2) 公式 2 =OFFSET(A2,1,,2,3)

例 2: 利用名称作为基点, 得到右图中红色字体区域公式:

=OFFSET(DATA,2,1,3,2) 公式结果\$B\$4:\$C\$6

由于公式中指定了最终结果的行数(3)和列数(2), 那么系统只能将 DATA 中的左上角单元格 (A2)作为引用基点, 并不是整个 DATA 区域。

公式等同于:

=OFFSET(A2,2,1,3,2) 公式结果 \$B\$4:\$C\$6

进一步理解: row, column, 不是指的目标位置, 而是相对于 reference 的取值。

Offset 理解图示 2

	A	B	C	D	E	F	G	H
1		基点	员工编号	员工姓名	性别	年龄		C2: C8偏移1列
2			A0711	张三	男	10		张三
3	5		A0795	李四	男	20		李四
4			A0733	王五	女	30		王五
5		2	B1234	赵六	男	40		赵六
6				张大	男	50		张大
7			A0123	刘二	女	60		刘二
8		偏移	B1123	田七	男	50		田七
9								=OFFSET (C2:C8, 0, 1)
10		公式	=OFFSET (B1, 5, 2, 3, 3)					
11		检验:	张大	男	50			
12			刘二	女	60			
13			田七	男	50			

1. 从以上单元格区域中取出来最右下角 9 个单元格, 放到新的 9 个单元格区域中。

拖动选择 3 行 3 列的单元格 > 输入=OFFSET(B1,5,2,3,3) > ctrl+shift+enter

说明:

从基点 B1 到 "张大" 所在单元格移动的行数为 5,列数为 2, 取出来 3 行 3 列的数据

2. 要得到姓名列包含标题行在内的所有数据

选中与姓名列行数相同的单元格区域 \Rightarrow `=OFFSET(B1,0,2,8,1)` \Rightarrow `ctrl+shift+enter`
也可以写为 `=OFFSET(B1,,2,8)`

说明: 第 2 个参数 0 可以省略, 最后 2 个参数即返回的单元格区域的行数及列, 如果与基点所在的行数和列数相同, 也可以省略, 在这里只有列数与参考点的列数相同, 都有 1 列, 故可以省略列数, 不能省略行数.

3. 把 D2:D8 员工姓名取出来

选中 7 个竖直单元格 \Rightarrow `=OFFSET(B1,1,2,7,1)`
由于此时目标区域与源区域的列数相同, 可以把参数 1 省略.

解释:

- 1, 通常情况下, 用户都是输入正数来控制产生引用的行数和列数, 偏移方向为向下取行数或向右取得列数。
- 2, 值得提醒的是, EXCEL 支持用户在行数_height 和列数_width 参数中输入负数来产生引用。负行数向上取行数, 负列数向左取列数。

Offset 应用: 对合并的单元格快速填充指定内容

姓名	课程	成绩
	数学	78
	计算机	100
	英语	88
	数据库	90
	数学	90
	计算机	90
	数学	90
	计算机	89
	英语	72
	语文	56
	化学	78
	数学	83
	计算机	78
	英语	76
	数学	75
	计算机	99

姓名
徐立特
阳叶萍
王双平
徐恒鹏
余世帆
谭高鑫

把已有的姓名列中的内容复制到已合并单元格的姓名列中

方法:

1. 选中 A 列从 A2 开始的合并单元格, 输入公式: =OFFSET(\$H\$1,COUNTA(\$A\$1:A1),)
2. ctrl+enter

解读:

1. COUNTA(\$A\$1:A1)获取当前 1 列中的单元格到\$A\$1 单元格已经有文本的单元格数目, 对于合并的单元格, 数据实际上是保存在第 1 个单元格中的, 其它单元格还是空白的单元格, 所以可以使用 COUNTA()来获取非空单元格的数目.
2. COUNTA 用来确定 offset 函数的行偏值, 上面已经有了 n 个非空的单元格, 就表示 已经填入了 n 个姓名, 从 H 列取第 n+1 个姓名填入到下一个合并的单元格中.

合并后的单元格无法使用填充柄来填充内容, 要选择所有合并后的单元格 A2:A17 > =OFFSET(\$H\$1,COUNTA(\$A\$1:A1),) > ctrl+enter

对选中的多个单元格中输入同样的数据时使用 ctrl+enter, 合并后的单元格中复制公式时也要使用 ctrl+enter

此时选择姓名列姓名所在的单元格为基点, 且修改为绝对引用, 在向下填充内容时, 偏移 1 个就取出第 1 个姓名, 偏移 2 个就取出第 2 个姓名, 所以基点一定要设置为绝对引用.

填入一个人的姓名之后, 因为姓名列中填入的是文本, 要使用 COUNTA 函数来确定偏移的行数, COUNTA(\$A\$1:A1)表示从 A1 单元格到当前单元格行的偏移量, 是用已经填写过姓名的数量来确定的.

在合并的单元格中填入的内容实际上是位于第一个单元格中的, 其余的单元格都是空值. 可以认为合并单元格也是对单元格格式的设置, 是显示方式的变化.

第 3 个参数是要取的数据与基点位置的列偏移量, 因为是从同一列中取数据, 所以没有偏移量, 只写一个逗号即可.

OFFSET 函数, 如果取出来的是值一个单元格, 就会返回其中的数据, 类似于查找的功能. 如果结果是多个单元格, 得到的是一个区域, 要先选择一个区域, 再用 ctrl+enter 来完成输入. 在上面的例子中, 取出来的结果就是一个单元格, 就把这个单元格的值取出来.

COUNTA 函数计算包含任何类型的信息(包括错误值和空文本 (""))的单元格。例如, 如果区域中包含的公式返回空字符串,COUNTA 函数计算该值。但 COUNTA 函数不会对空单元格进行计数。

##如果不需要对逻辑值, 文本或错误值进行计数(换句话说,只希望对包含数字的单元格进行计数), 请使用 COUNT 函数。

##如果只希望对符合某一条件的单元格进行计数, 请使用 COUNTIF 函数或 COUNTIFS 函数。

公式说明结果

=COUNTA(A2:A7)计算单元格区域 A2 到 A7 中非空单元格的个数。返回的结果为 5
计算时包含 A2 和 A7 在内。

Offset 应用：累积和的计算

在 C 列计算 B 列的累积数据。

可以使用填充柄拖动, 也可以选择所要区域, 输入函数后, ctrl+enter 应用于所有单元格

=SUM(OFFSET(B\$2,0,0,ROW()-1,1))

也可以把基点选择在 B1 单元格上, =SUM(OFFSET(\$B\$1,1,0,ROW()-1,1))

使用 OFFSET 函数来确定对哪部分数据进行加和运算。

参考点定位在 293 也就是第一个数据上, 因为每一次都是从 293 开始进行累加运算, 所以要使用绝对引用, 且向下向右移动 0 行 0 列, 相当于还是从 B\$2 单元格开始取值。

对 1, 2, 3,..., 个单元格的累计, OFFSET 的第 4 个参数控制高度的变化, 在不同行中是变化的, 所以 offset 函数的第 3 个参数 height 即行高是变化的。

表示从 B\$2 开始取出 ROW()-1 个数据, 以数组的形式返回到 SUM 函数中。

第二行的累加只加 1 个数字, 第三行的累加加 2 个数字... 如此就实现了累加数据计算。

使用公式求值来理解上面的过程

数据是保存在工作表的不同区域中的, 为了取出来某一区域的数据或者批量向某一区域中填入数据, 就可以使用函数来动态批量的完成。借助于函数对行或列中的数据动态的获取或填入。

MATCH

给了值找位置

功能: 在单元格区域中搜索指定项, 然后返回该在单元格区域的相对位置。

MATCH(lookup_value, lookup_array, [match_type])

例如, 如果单元格区域 A1:A3 包含值 5, 25 和 38 , 则以下公式:

=MATCH(25,A1:A3,0)

会返回数字 2, 因为值 25 是单元格区域中的第二项。

match_type 可选 :

-1: 查找大于或等于 lookup_value 的最小值 ,参数中的值必须按降序排列

0 :查找等于 lookup_value 的第一个值

1 (默认值)查找小于或等 lookup_value 的最大值 ,参数中的值必须按升序排列

使用特点

第 2 个参数: 1 行或 1 列数组

返回一个数, 表示查找值的行号或列号

常用在

Index 函数的行号或列号

OFFSET 的参数(偏移量等)

区别(解决同一个问题)

=INDEX(A1:D11,MATCH(F1,A1:A11,0),MATCH(G1,A1:D1,0))

=OFFSET(A1,MATCH(F1,A2:A11,0),MATCH(G1,B1:D1,0))

Match 函数帮助信息

使用 MATCH 函数在 范围 单元格中搜索特定的项, 然后返回该项在此区域中的相对位置。

例如, 如果 A1:A3 区域中包含值 5, 25 和 38 , 那么公式 =MATCH(25,A1:A3,0) 返回数字 2, 因为 25 是该区域中的第二项。

提示: 当您需要项目在区域中的位置而非项目本身时, 使用 MATCH 而不是 LOOKUP 函数之一。例如, 您可以使用 MATCH 函数提供 INDEX 函数的 row_num 参数值。

语法

MATCH(lookup_value, lookup_array, [match_type])

MATCH 函数语法具有下列参数:

lookup_value 必需。要在 lookup_array 中匹配的值。例如, 如果要在电话簿中查找某人

的电话号码，则应该将姓名作为查找值，但实际上需要的是电话号码。

lookup_value 参数可以为值(数字，文本或逻辑值)或对数字，文本或逻辑值的单元格引用。

lookup_array 必需。要搜索的单元格区域。

match_type 可选。数字 -1, 0 或 1。**match_type** 参数指定 Excel 如何将 **lookup_value** 与 **lookup_array** 中的值匹配。此参数的默认值为 1。

下表介绍该函数如何根据 **match_type** 参数的设置查找值。

Match_type

行为

1 或省略

MATCH 查找小于或等于 **lookup_value** 的最大值。**lookup_array** 参数中的值必须以升序排序，例如：...-2, -1, 0, 1, 2, ..., A-Z, FALSE, TRUE。

0

MATCH 查找完全等于 **lookup_value** 的第一个值。**lookup_array** 参数中的值可按任何顺序排列。

-1

MATCH 查找大于或等于 **lookup_value** 的最小值。**lookup_array** 参数中的值必须按降序排列，例如：TRUE, FALSE, Z-A, ...2, 1, 0, -1, -2, ... 等等。

MATCH 返回匹配值在 **lookup_array** 中的位置，而非其值本身。例如，**MATCH**("b",{"a","b","c"},0)返回 2，即"b"在数组 {"a","b","c"} 中的相对位置。

匹配文本值时，**MATCH** 函数不区分大小写字母。

如果 **MATCH** 函数查找匹配项不成功，它会返回错误值 #N/A。

如果 **match_type** 为 0 且 **lookup_value** 为文本字符串，您可在 **lookup_value** 参数中使用通配符 - 问号 (?) 和星号 (*)。问号匹配任意单个字符；星号匹配任意一串字符。如果要查找实际的问号或星号，请在字符前键入波浪符 (~)。

视频演示

MATCH 理解

给出值返回行号或列号

你是老几?

MATCH(lookup_value, lookup_array, [match_type])

lookup_value 如果是数字,就不需要加引号,如果是文本,就要加引号.

功能:给出一个查找值,返回在指定范围的相对行号或列号。

MATCH 函数在范围单元格中搜索特定的项,然后返回该项在此区域中的相对位置.

只能对一列或一行进行查找

0:任何排序

1:升序

-1:降序

查找位置--行号:

B

1

2

3003

4

5

6

1007

8

9

10

11

12

13

查询 100 这个值在一列数据中的行号

=MATCH(100,B2:B14,0)

查找数字时可以不加双引号. B2:B14 为查找的范围, 并且是精确查找.

查找位置--列号:

1000 中国 200

1234567

查找 200 在一行数据中所处的列号

=MATCH(200,E4:K4,0)

MATCH 函数的用处: 相同个数

作业:计算两列数据相同数目。

数据 1 数据 2

left1rand

rightBug

midthanKs

roundright

randmid

Bugr0und

thanksmonny

monnytext

testleft

Afterafter

共 10 个单词

解析:

MATCH(A2:A11,B2:B11,0)

0 表示精确匹配, 1 表示小于, -1 表示大于

结果是 {9;4;5;#N/A;1;2;3;7;#N/A;10}

COUNT(MATCH(A2:A11,B2:B11,0)) > CTRL+SHIFT+ENTER

表示第一列的数据分别在第 2 列的位置行号,若没有,则以#N/A 表示,统计时不计#N/A 数,

不区分大小写,A 列是查找值, 在 B 列是查找范围

使用 F9 或公式求值来查看过程

标记匹配结果

查找学生姓名

酆六一(B1)魏三○殷七四卫八尤一九顾九三褚一一

第 1 组(B3)曹二六葛四○凤五四卫八赵一杨一二余九○陶三一奚四五

第 2 组(B4)严二七云四一花五五倪七一钱二蒋一三元九一喻三二范四六

第 3 组(B5)姜二八苏四二柳五六毕七二孙三沈一四黄九二戚三三彭四七

第 4 组(B6)金二九潘四三俞五七滕七三王四韩一五顾九三谢三四马四八

第 5 组(B7)魏三○郎四四任五八殷七四周五许一六孟九四邹三五鲁四九

第 6 组(B8)陶三一奚四五袁五九罗七五吴六朱一七平九五章三六韦五○

第 7 组(B9)喻三二范四六唐六○常七六郑七秦一八尹九六葛四○凤五四

第 8 组(B10)戚三三彭四七酆六一郝七七卫八尤一九和九七云四一花五五

第 9 组(B11)谢三四马四八鲍六二邬七八冯九张二○穆九八苏四二柳五六

第 10 组(B12)邹三五鲁四九史六三安七九陈一○何二一萧九九潘四三俞五七

第 11 组(B13)章三六韦五○薛六四傅八○褚一一吕二二汪一○○傅八○褚一一

在 99 个人中去查找特定的 7 个人, 并把结果高亮显示

条件格式的本来用处是把图表搬到单元格中

选中 99 个人所在的单元格区域 > 开始 > 样式 > 条件格式 > 清除规则 > 清除所选单元格的规则

选中 99 个人所在的单元格区域 > 开始 > 样式 > 条件格式 > 新建格式规则 > 选择规则类型:使用公式确定要设置格式的单元格 > 编辑规则说明:=MATCH(B3,\$B\$1:\$H\$1,0) > 格式 > 设置格式, 如在填充中选择一个颜色

说明:

B3 为查找目标, 是相对引用地址, 函数会自动扩充到 B3 所在单元格向右向下的所有区域的行和列.

查找范围为 7 个学生的名字, 因为要扩充到其它单元格中, 所以要锁定\$B\$1

如果找到了值, 就给它们设定新的格式

对结果的理解:

选择与 B3:J13 相同大小的区域 > 输入=MATCH(B3:J13,B1:H1,0)

也就是在所有数据中查找特定姓名的几个人, 返回查找到的姓名的特定位置.

在逻辑运算中 false=0, 凡是不为 0 的数, 都被认为是 true.

选中原数据的所有单元格 > 开始 > 样式 > 条件格式 > 管理规则中查看在条件格式中运用了设置, 但是没有设置等于多少, 凡是不为 0 的都为 true, 这样就能设置查找到的结果的格式了.

INDEX 函数

INDEX 函数语法功能

给了位置找值

功能: 返回单元格或数组中的数据或元素值, 此元素由行号和列号的索引来给定。

INDEX(array,row_num,[column_num])

INDEX(reference,row_num,[column_num],[area_num])

Array —— 单元格区域或数组常量。

Row_num—— 数组中某行的行号，函数从该行返回值。如果省略 row_num，则必须有 column_num。

Column_nu—— 数组中某列的列标，函数从该列返回值。如果省略 column_num，则必须有 row_num。

INDEX 使用特点

给出选定区域的行号或列，获取查找值

多行多列

单行

单列

常用在

指定行或列的查找

INDEX+MATCH 代替 VLOOKUP

INDEX 函数帮助信息

INDEX 函数返回表格或区域中的值或值的引用。

使用 INDEX 函数有两种方法：

- 如果想要返回指定单元格或单元格数组的值，请参阅[数组形式](#)。
- 如果想要返回对指定单元格的引用，请参阅[引用形式](#)。

数组形式

说明

返回表格或数组中的元素值，此元素由行号和列号的索引值给定。

当函数 INDEX 的第一个参数为数组常量时，使用数组形式。

语法

INDEX(array, row_num, [column_num])

INDEX 函数语法具有下列参数：

Array 必需。单元格区域或数组常量。

如果数组只包含一行或一列，则相对应的参数 Row_num 或 Column_num 为可选参数。

如果数组有多行和多列，但只使用 Row_num 或 Column_num，函数 INDEX 返回数组中的整行或整列，且返回值也为数组。

Row_num 必需。选择数组中的某行，函数从该行返回数值。如果省略 Row_num，则必须有 Column_num。

Column_num 可选。选择数组中的某列，函数从该列返回数值。如果省略 Column_num，则必须有 Row_num。

备注

- 如果同时使用参数 Row_num 和 Column_num，函数 INDEX 返回 Row_num 和 Column_num 交叉处的单元格中的值。
- 如果将 Row_num 或 Column_num 设置为 0（零），函数 INDEX 则分别返回整个列或行的数组数值。若要使用以数组形式返回的值，请将 INDEX 函数以数组公式形式输入，对于行以水平单元格区域的形式输入，对于列以垂直单元格区域的形式输入。若要输入数组公式，请按 Ctrl+Shift+Enter。

注意：在 Excel Web App 中，不能创建数组公式。

- Row_num 和 Column_num 必须指向数组中的一个单元格；否则，INDEX 返回错误值 #REF!。

示例

示例 1

下面的示例使用 INDEX 函数查找某一行和某一列的交叉单元格中的值。

复制下表中的示例数据，然后将其粘贴进新的 Excel 工作表的 A1 单元格中。

要使公式显示结果，请选中它们，按 F2，然后按 Enter。如果需要，可调整列宽以查看所有数据。

数据	数据
苹果	柠檬
香蕉	梨
公式	说明
=INDEX(A2:B3,2,2)	位于区域 A2:B3 中第二行和第二列交叉处的数值。
=INDEX(A2:B3,2,1)	位于区域 A2:B3 中第二行和第一列交叉处的数值。

示例 2

此示例在数组公式中使用 INDEX 函数查找一个 2x2 数组中指定两个单元格中的值。

在此例中，选择任意两个垂直单元格，如 A1:A2，将下列公式
=INDEX({1,2;3,4},0,2) 粘贴到第一个单元格中，然后按
CTRL+SHIFT+ENTER。当作为数组公式输入时，Excel 会自动将公式粘贴到第
二个单元格中。

公式	说明	结果
=INDEX({1,2;3,4},0,2)	数组的第一行、第二列中找到的数值。数组包含第一行中的 1 和 2 以及第二行中的 3 和 4。	2
	数组（与上面的数组相同）的第二行、第二列中找到的数值。	4

注意：这是数组公式，需使用 Ctrl+Shift+Enter 输入。Excel 会自动将公式括在大括号 {} 中。如果尝试自己输入，Excel 会将公式作为文本显示。如果不使用 Ctrl+Shift+Enter，则公式将返回一个 #VALUE! 错误。

引用形式

说明

返回指定的行与列交叉处的单元格引用。如果引用由不连续的选定区域组成，可以选择某一选定区域。

语法

INDEX(reference, row_num, [column_num], [area_num])

INDEX 函数语法具有下列参数：

- **Reference** 必需。对一个或多个单元格区域的引用。
 - 如果为引用输入一个不连续的区域，必须将其用括号括起来。
 - 如果引用中的每个区域只包含一行或一列，则相应的参数 Row_num 或 Column_num 分别为可选项。例如，对于单行的引用，可以使用函数 INDEX(reference,,column_num)。
- **Row_num** 必需。引用中某行的行号，函数从该行返回一个引用。
- **Column_num** 可选。引用中某列的列标，函数从该列返回一个引用。
- **Area_num** 可选。在引用中选择要从中返回 Row_num 和 Column_num 的交叉处的区域。选择或输入的第三个区域编号为 1，第二个为 2，依此类推。如

果省略 Area_num，则 INDEX 使用区域 1。此处列出的区域必须全部位于一张工作表。如果指定的区域不位于同一个工作表，将导致 #VALUE! 错误。如果需要使用的范围彼此位于不同工作表，建议使用函数 INDEX 的数组形式，并使用其他函数来计算构成数组的范围。例如，可以使用 CHOOSE 函数计算将使用的范围。

例如，如果引用描述的单元格为 (A1:B4,D1:E4,G1:H4)，则 Area_num 1 为区域 A1:B4，Area_num 2 为区域 D1:E4，而 Area_num 3 为区域 G1:H4。

备注

- Reference 和 Area_num 选择了特定的区域后，Row_num 和 Column_num 将进一步选择特定的单元格：Row_num 1 为区域的首行，Column_num 1 为首列，以此类推。函数 INDEX 返回的引用即为 Row_num 和 Column_num 的交叉区域。
- 如果将 Row_num 或 Column_num 设置为 0，函数 INDEX 分别返回对整列或整行的引用。
- Row_num、Column_num 和 Area_num 必须指向 reference 中的单元格；否则，INDEX 返回 错误值 #REF!。如果省略 Row_num 和 Column_num，函数 INDEX 返回由 Area_num 所指定的引用中的区域。
- 函数 INDEX 的结果为一个引用，且在其他公式中也被解释为引用。根据公式的需要，函数 INDEX 的返回值可以作为引用或是数值。例如，公式 `CELL("width",INDEX(A1:B2,1,2))` 等价于公式 `CELL("width",B1)`。CELL 函数将函数 INDEX 的返回值作为单元格引用。而在另一方面，公式 `2*INDEX(A1:B2,1,2)` 将函数 INDEX 的返回值解释为 B1 单元格中的数字。

注意：Excel Web App 中 CELL 函数不可用。

示例

复制下表中的示例数据，然后将其粘贴进新的 Excel 工作表的 A1 单元格中。要使公式显示结果，请选中它们，按 F2，然后按 Enter。如果需要，可调整列宽以查看所有数据。

水果	价格	计数
苹果	¥6.90	40

水果	价格	计数
香蕉	¥3.40	38
柠檬	¥5.50	15
橙子	¥2.50	25
梨	¥5.90	40
杏	¥28.00	10
腰果	¥35.50	16
花生	¥12.50	20
胡桃	¥17.50	12

公式	说明	结果
=INDEX(A2:C6, 2, 3)	区域 A2:C6 中第二行和第三列的交叉处，即单元格 C3 的内容。	38
=INDEX((A1:C6, A8:C11), 2, 2)	第二个区域 A8:C11 中第二行和第二列的交叉处，即单元格 B9 的内容。	1.25
=SUM(INDEX(A1:C11, 0, 3, 1))	对第一个区域 A1:C11 中的第三列求和，即对 C1:C11 求和。	216
=SUM(B2:INDEX(A2:C6, 5, 2))	返回以单元格 B2 开始到单元格区域 A2:A6 中第五行和第二列交叉处结束的单元格区域的和，即单元格区域 B2:B6 的和。	

视频演示

INDEX 理解

INDEX(array,row_num,[column_num])

返回单元格或数组中的数据或元素值，此元素由行号和列号的索引值来给定

特点:给出行或列号，在指定范围获取查找值。
可以单独查找一行或一列的数据，也可以在一个单元格区域中查找指定列号和行号的值。

查找行号为 3，列号为 2 的值

姓名(B1)部门补助
花花交通部 1000
兰兰安监部 1200
玉玉储运部 1500
茵茵卫生部 1600
丫丫法务部 1300

=INDEX(B2:D6,3,2) 结果:储运部

在补助列中查找第 3 行中的数据

=INDEX(D2:D6,3)

查找丫丫的补助

=INDEX(B6:D6,3)

INDEX 使用实例:提取姓名

作业:利用函数 INDEX+ROW 提取姓名。

学号(A1)5947(B1)

姓名赵一

学号 1784

姓名钱二

学号 8273

姓名孙三

学号 8398

姓名李四

学号 7866

姓名周五

学号 6637

姓名吴六

姓名

=INDEX(\$B\$1:\$B\$12,(ROW()-1)*2)

因为要复制到下面的单元格中, 所以要对区域进行绝对引用.

ROW(A1)--带参数

也可以使用

=INDEX(B:B,(ROW()-1)*2)

B:B 也是把引用固定在了 B 列, 并且向下拖动填充柄时引用不会发生移动.

思考: 为什么取 B:B,或使用绝对引用方可?

学号在奇数行, 姓名在偶数行, 要提取姓名, 如何实现?

函数的难点:1.如何选择函数,2 函数的参数和位置

进阶-动态计算学生总成绩

姓名语文数学 英语总成绩

姚良德 102.5103106.5

宋国爱 115105.5138.5

刘猛 122.5145.5135.5

郑弟琴 138141133.5

宋建 84.5143.5129.5

岑仕廷 129.5130131.5

方井田 119.5145.5107

李兴义 127.597114.5

罗朝周 138141110

杨昌艳 147.5106147.5

要求:利用 SUM,OFFSET 和 MATCH 函数同姓名获取对应成绩总和的功能

每个学生总成绩

姓名总成绩(3 门之和)用 E 列数据

宋国爱#REF!

选中姓名下面的单元格 > 数据 > 数据工具 > 数据验证 > 序列 > 选择 A2:A11

可以先在 E 列求出总成绩,再在数据验证中进行引用

也可以直接引用 BCD 三列的成绩求出来总成绩

=SUM(OFFSET(B1,MATCH(A16,A2:A11,0),,3))

这里 OFFSET 函数的高度参数必须要省略,不能使用 0,3,来表示,否则会出错.

因为相加的数据是从 B2 开始的,所以把 OFFSET 函数的基点定位在 B1 单元格中,如果行偏是 1,就取到了第 1 个人的成绩,如果行偏是 2,就取到了第 2 个人的成绩.

使用 MATCH 函数来获取 OFFSET 函数的行偏,

=MATCH(A16,A2:A11,0)

得出 A16 单元格中的名称在 A2:A11 这个序列中的序号

在 A16 单元格中选择其他人,MATCH 函数的值就发生相应的变化.这个值就作为 OFFSET 函数的行偏,就可以在姓名中取到对应的值了.所以 MATCH 函数经常和 OFFSET 函数或 INDEX 函数配合使用.

在使用 MATCH 函数得出来偏移量时,由于 OFFSET 的偏移量是从基点开始计算的,基点并不包含在内,所以要把基点设定在 B1 单元格上.

要求:动态计算产品价格。

Index 和 match 函数

型号/规格 10(E1)2030

A0110(D2)788776

A0111809784

A0112917564

A0113888668

A0114939983

B1120896979

B1121917069

B1122779181

B1123987774

信息查询(A2)

查询型号(A3)

查询规格

产品价格

查询型号 B3 > 数据 > 数据验证 > 允许:序列, 来源:D2:D10

查询规格 B4 > 数据 > 数据验证 > 允许:序列, 来源:E1:G1

使用 match 函数来返回行号和列号, 使用 Index 函数来根据行号和列号查找数据

行号为 MATCH(B3,D2:D10,0)

列号为 MATCH(B4,E1:G1,0)

产品价格: B5 =INDEX(E2:G10,MATCH(B3,D2:D10,0),MATCH(B4,E1:G1,0))

OFFSET+MATCH

姓名(A1)语文数学英语总成绩(E1)

姚良德 102.5103106.5312

宋国爱 115105.5138.5359

刘猛 122.5145.5135.5403.5

郑弟琴 138141133.5412.5

宋建 84.5143.5129.5357.5

岑仕廷 129.5130131.5391

方井田 119.5145.5107372

李兴义 127.597114.5339

罗朝周 138141110389

杨昌艳 147.5106147.5401

每个学生总成绩
姓名(A14)总成绩

A15 中使用序列生成所有人的姓名,选中了某个人的姓名,原数据中这个人所在的行就高亮显示.

使用 MATCH 函数来计算行偏移量

条件格式: =\$A2=\$A\$15

A2 锁定了列,没有锁定行,它可以自动扩展到行中,也就是条件格式会应用到查找到单元格所在的一整行中,而不是只应用于查找到的一个单元格.

如果条件格式中的公式改为 =A2=\$A\$15,那么条件格式只会应用到查找到的单元格中,不会扩展到整行中.

A15 为通过数据验证得到的学生的姓名,行和列都锁定.以保证自动扩充到其它单元格时不会发生变化.

VLOOKUP

函数解释:

搜索表区域首列满足条件的元素,确定待检索单元格所在的区域中的行序号,再进一步返回选定单元格的值。

语法: VLOOKUP(lookup_value,table_array,col_index_num,range_lookup)

1. lookup_value 为需要在表格首列中查找的数值。
2. table_array 为需要在其中查找数据的单元格区域。
3. col_index_num 为 table_array 中待返回的匹配值列序号。
4. range_lookup 为一逻辑值,指明函数 VLOOKUP 返回时是精确匹配还是近似匹配。如果为 TRUE 或省略,则返回近似匹配值也就是说如果找不到精确匹配值,则返回小于 lookup_value 的最大数值;如果 range_value 为 FALSE,函数 VLOOKUP 将返回精确匹配值。如果找不到,则返回错误#N/A。

您需要四条信息才能构建 VLOOKUP 语法:

1. 要查找的值,也被称为查阅值。
2. 查阅值所在的区域。请记住,查阅值应该始终位于所在区域的第一列,这样 VLOOKUP

才能正常工作。例如，如果查阅值位于单元格 C2 内，那么您的区域应该以 C 开头。

- 3. 区域中包含返回值的列号。例如，如果指定 B2:D11 作为区域，那么应该将 B 算作第一列，C 作为第二列，以此类推。
- 4. 可选。如果需要返回值的近似匹配，可以指定 TRUE；如果需要返回值的精确匹配，则指定 FALSE。如果没有指定任何内容，默认值将始终为 TRUE 或近似匹配。

现在将上述所有内容集中在一起，如下所示：

=VLOOKUP(查阅值, 包含查阅值的区域, 区域中包含返回值的列号以及(可选)为近似匹配指定 TRUE 或者为精确匹配指定 FALSE)。

=VLOOKUP(要查找的值, 要在其中查找值的区域, 区域中包含返回值的列号, 精确匹配或近似匹配-指定为 0/FALSE 或 1/TRUE)。

=VLOOKUP(找啥,在哪块儿找,找到后要这块区域中哪一列的内容,是精确查找还是近似查找)

VLOOKUP 示例:

姓名	部门
花花	交通部
兰兰	安监部
玉玉	储运部
苗苗	卫生部
丫丫	法务部

姓名	部门
玉玉	储运部
=VLOOKUP(A9, A1:B6, 2, 0)	

公式的含义：从 A1:B6 单元格区域中精确查找 A9 单元格中的值，并返回 A1:B6 单元格区域中与查找到的值对应的第 2 列的值。

实例 2：让函数返回符合条件的多条记录

姓名	成绩	等级	辅助列
聂倩	127	及格	
张成艳	100	不及格	
王芸	77	不及格	
孙中心	142	优秀	
张志成	112	不及格	
王勇	140	优秀	
顾成	60	不及格	

阮永	138	及格	
卢林涛	127	及格	
阵志琴	91	不及格	
梁守印	85	不及格	
郑少红	131	及格	
孟俊	69	不及格	
彭继伟	136	不及格	

辅助列: =COUNTIF(A\$2:A2,\$E\$2&"*")

查找:=VLOOKUP(ROW(1:1),\$A:\$D,COLUMN(B:B),FALSE)

多个条件

第一个参数: A2&B2

????

视频演示

VLOOKUP 实例 1

作业:提取数据库成绩。

每个老师都提供一个科目的成绩, 要把一个学生所有科目的成绩汇总到一起.

原数据:

学号(G2)姓名数据库技术与编程

20111043102 谭高鑫 82.8

20111043103 宋志阳 59.2

20111043104 王千慧 68

20111043105 高艺 68.8

20111043106 蔡仲璋 85.8

20111043107 李文琪 88.4

学号(A2)姓名大学计算机基础数据库技术与编程

20111043109 黄舒馨 52

20111043117 王嵘 52

20111043206 杨宁 55

20111043116 邹宛廷 56

20111043123 徐叶 56

20111043131 王风 56

要从原数据中取出来"数据库技术与编程"课程的成绩. 原数据和结果数据中学生的排列次序可能不一样.

=VLOOKUP(A3,\$G\$3:\$I\$123,3,FALSE)

表示从原数据库技术与编程成绩单中精确查找学号为 A3 的学生,把第 3 列的数据也就是"数据库技术与编程"的成绩数据返回到本单元格中.
因为要复制到其它的单元格中,要把查找区域锁定

在 B1:C6 区域中查找名为玉玉的结果,找玉玉第二个单元格中的内容作为结果放在本单元格中

实例 2

省份简称 省会 省花 景点 特产 区号
江西(C14) 赣 南昌 杜鹃 庐山瓷器 0791
浙江(C15) 浙 杭州 玉兰 西湖龙井茶 0571
江苏(C16) 苏 南京 茉莉 苏州园林 苏绣 0512
四川(C17) 蜀 成都 木芙蓉 九寨沟榨菜 028
安徽(C18) 皖 合肥 紫薇 黄山徽墨 0551
湖南(C19) 湘 长沙 荷花 张家界臭豆腐 0731
1234567

=VLOOKUP(D8,C14:I19,F7,0)

第 1 参数,查找的内容

D8=INDEX(C14:C19,7-C8)

查找的内容是省份, 选择区域时要把省份放在所选区域的最左列中.

如果查找的内容是区号, 就要把区号放在最左列中

第 2 参数,查找区域

如果要复制到其它的区域中, 必须要把查找区域锁定.

第 3 参数,返回查找区域内第几列的数据

F7=INDEX(C13:I13,F7)&"

第 4 参数,匹配(精确或模糊)

公式解释

= "公式解释: 在 C14:I19 范围内查找首列等于 "&D8&" 对应第 "&F7&" 列的值。结果为:"&I8

使用 VLOOKUP 函数从身份证信息中查找省市信息

员工档案表

编号姓名性别身份证号码入职时间出生日期年龄工龄(年)以及不足一年的工龄(月)所属省自治区直辖市所属区县职务最高学历所属部门联系电话

CYU001 李楠男 3722291980032312152003/1/23 主管本科生产部 83983474

CYU002 方鹏男 1410821978101321342000/6/24 副主管本科生产部 83983124

CYU003 李磊男 3501221979112018111999/7/23 员工本科生产部 83983648

CYU004 王小若女 3501241980081018412001/12/26 主管硕士管理部 83988577

CYU005 陈雨女 1403031974050623201993/4/20 副主管博士管理部 83984869

CYU006 石璐女 1403031977051104621995/8/28 员工其他管理部 83984275

CYU007 张瑛女 1403031984050623202005/9/9 员工硕士管理部 83987597

CYU008 程晓女 1403031974031104421997/3/30 主管其他销售部 83989542

CYU009 王丽女 1403031978050623822000/5/31 主管博士销售部 83934678

CYU010 赵军力男 3501231973112018111990/2/1 员工大专销售部

在全国地区表工作表中选择所有的地区数据 > 公式 > 定义的名称 > 定义名称 > 省市列表

使用自定义名称来预防可能出现的地址引用的绝对或相对的变换问题

所属省自治区直辖市,可由身份证前 2 位得出

=VLOOKUP(LEFT(D2,2),省市列表,2,FALSE) > 双击填充柄

表示从定义的省市列表中精确查找身份证前 2 位,并把查找到的省市列表中第 2 列的数据返回.

如果不使用省市列表的自定义名称,可以写为

=VLOOKUP(LEFT(D2,2),全国地区表!\$A\$1:\$B\$34,2,FALSE)

所属区县,由身份证前 6 位得出

=VLOOKUP(LEFT(D2,6),省市列表,2,FALSE)

使用数组的方法来完成

有多少个身份证号,就要选中与之对应的多少个单元格 > =VLOOKUP(LEFT(D2:D25),2,省市列表,2,0) > ENTER

个人所得税

作业:使用 vlookup 计算个人所得税。

个人所得税税率表(工资,薪金所得适用)起征点金额:¥3,500.00

级数(B2)全月应纳税所得额税率(%) (D2)税率(E2)速算扣除数(F2)

- 1 不超过 1500 元的 00.050
- 2 超过 1500 元至 4500 元的部分 15000.175
- 3 超过 4500 元至 9000 元的部分 45000.2525
- 4 超过 9000 元至 35000 元的部分 90000.25975
- 5 超过 35000 元至 55000 元的部分 350000.32725
- 6 超过 55000 元至 80000 元的部分 550000.355475
- 7 超过 80000 元的部分 800000.4513475

速算扣除数

如 4500-9000 的部分,假如工资为 10000, $10000-3500=6500$, 应纳税部分为 6500, 在 4500-9000 之间, 可以把 6500 全部按税率 0.2 来计算, 得出来 $6500*0.2=1300$, 因为多算了, 所以要减去速算扣除数中的值. $1300-75=1225$, 应纳税 1225

编号(H2)纳税人姓名月收入(J2)应纳税收入(K2)个人税(L2)实发工资(M2)

- 1 杨一二 23,092.43 19,592.43
- 2 蒋一三 10,727.55 7,227.55
- 3 沈一四 9,000.00 5,500.00
- 4 韩一五 5,108.15 1,608.15
- 5 许一六 7,000.75 3,500.75
- 6 朱一七 18,513.15 15,013.15
- 7 秦一八 3,450.00 0.00
- 8 尤一九 19,085.35 15,585.35
- 9 张二〇 2,966.95 0.00
- 10 何二一 12,371.15 8,871.15

应纳税收入 K3

=IF(J3<=3500,0,J3-3500)

或者

=IF(J3<=3500,0,J3-\$F\$1)

F1 为起征点金额 3500

个人所得税

选中个人税下面的单元格 L3

=VLOOKUP(K3,\$D\$3:\$F\$9,2,1)*K3-VLOOKUP(K3,\$D\$3:\$F\$9,3,1)

在税率表 D3:F9 中查找收入中应纳税的部分 K3 处在哪个区域中, 返回查找区域中的第 2 列, 即税率的列, 因为是范围的查找, 所以要用 1 模糊查找. 减去速算扣除数中对应的数值.

使用数组来完成

选中个人税下面所有单元格 L3:L12 > =K3:K12*VLOOKUP(K3:K12,D3:F9,2,TRUE)-
VLOOKUP(K3:K12,D3:F9,3,TRUE) > ctrl+shift+enter

说明:

应纳税收入*税率-多缴纳的部分

课后作业-HLOOKUP 查询

幸福媒体传播公司

十一月份 员工薪资表

姓名(A3)业绩底薪工龄工龄工资应发薪水

苏玫洁¥450,000¥0

罗丹伶¥500,000¥0

颜世璇¥800,000¥0

苏俊明¥1,000,000¥0

王思涵¥2,500,000¥0

业绩(A10)¥0¥500,000¥1,000,000¥2,000,000¥3,000,000

底薪(A11)¥10,000¥15,000¥20,000¥25,000¥30,000

工龄(A13)012345

工龄工资(A14)¥1,000¥1,500¥2,000¥2,500¥3,000¥4,000

HLOOKUP 函数与 VLOOKUP 函数刚好相反,只要把要查询的数据进行转置即可

底薪

选中底薪中的所有单元格 > =HLOOKUP(B4:B8,B10:F11,2,TRUE) > ctrl+shift+enter

查找值为 B4:B8,即员工的真实业绩,查找范围为业绩和底薪的对应关系值,需要得到的结果是业绩和底薪对应值中的底薪,即查找范围内的第 2 行,如果是对数字范围的查找,就使用模糊匹配.

总体的意思就是从业绩和底薪的对应关系中模糊查找具体的业绩所对应的业绩范围,并把业绩对应的底薪返回到当前单元格中.

工龄工资

定位到工龄工资中的第一个单元格 > =HLOOKUP(D4,\$B\$13:\$G\$14,2,FALSE) > 拖动填充柄填充到下面几个单元格

注意要对查找的范围使用绝对引用,否则在拖动填充柄时查找的范围就会向下移动,就会出错.

行，列标号函数

ROW([reference])

功能:返回引用的行号。

Reference —— 需要得到其行号的单元格或区域。

COLUMN ([reference])

功能:返回引用的列号。

Reference —— 需要得到其列号的单元格或单元格区域。

使用特点：

变化的特点: 作为随行/列变化的量

= COLUMN(), =ROW()

= COLUMN (A1), =ROW(A1)

= COLUMN (A:G), =ROW(4:10)

INDIRECT 函数

INDIRECT(ref_text,A1)

功能: 返回由文本字符串指定的引用。

Ref_text: 为对单元格的引用，此单元格可以包含 A1 -样式的引用， R1C1 -样式的引用，定义为引用的名称或对文本字符串单元格的引用。如果 ref_text 不是合法的单元格引用，函数 INDIRECT 返回错误值 #REF! 。

A1: 为一逻辑值，指明包含在单元格 ref_text 中的引用类型。

如果 a1 为 TRUE 或省略， ref_text 被解释为 A1 -样式的引用。

如果 a1 为 FALSE, ref_text 被解释为 R1C1 -样式的引用。

INDIRECT 函数帮助信息

本文介绍 Microsoft Excel 中 INDIRECT 函数的公式语法和用法。

说明

返回由文本字符串指定的引用。此函数立即对引用进行计算，并显示其内容。如果需要更改公式中对单元格的引用，而不更改公式本身，请使用函数 INDIRECT。

语法

INDIRECT(ref_text, [a1])

INDIRECT 函数语法具有以下参数:

Ref_text 必需。对单元格的引用,此单元格包含 A1 样式的引用,R1C1 样式的引用,定义为引用的名称或对作为文本字符串的单元格的引用。如果 ref_text 不是合法的单元格引用,则 INDIRECT 返回 错误值。

如果 ref_text 是对另一个工作簿的引用(外部引用),则被引用的工作簿必须已打开。如果源工作簿没有打开,则 INDIRECT 返回错误值 #REF!。

注意 Excel Web App 中不支持外部引用。

如果 ref_text 引用的单元格区域超出 1,048,576 这一行限制或 16,384 (XFD) 这一列限制,则 INDIRECT 返回错误 #REF!。

注意 此行为不同于 Microsoft Office Excel 2007 之前的 Excel 版本,早期的版本会忽略超出的限制并返回一个值。

A1 可选。一个逻辑值,用于指定包含在单元格 ref_text 中的引用的类型。

如果 a1 为 TRUE 或省略,ref_text 被解释为 A1-样式的引用。

如果 a1 为 FALSE,则将 ref_text 解释为 R1C1 样式的引用。

4 INDIRECT 的 R1C1 样式

数据 结果公式

A3A3=INDIRECT("A2")

A3=INDIRECT("R2C1",FALSE)

相同的结果,不同表达

INDIRECT 第二参数省略不写或为 TRUE(1)时为 A1 样式,第一参数表达时,要遵循 A1 样式的写法;

INDIRECT 第二参数为 FALSE(0)时为 R1C1 样式,第一参数表达时,要遵循 R1C1 样式的写法

INDIRECT 理解

INDIRECT 为间接引用。

使用&符号可以拼接文本地址,包括工作表的名称和单元格区域的名称。但是拼接之后是文

本的格式即有引号的文本,使用 INDIRECT 函数可以把引号去掉,变成 Excel 可以识别的引用.

数据(A1)函数显示结果

A3=INDIRECT(A2)100

100=INDIRECT("A2")A3

1.参数带引号的区别

参数不加引号,直接引用单元格数据

参数加引号,参数为文本,直接引用文本所代表的单元格。

2.使用不同工作表的数据(翻墙功能)

提取姓名(A10)个人所得税(B10)

姓名=INDIRECT(B\$10&"!I"&ROW(B2))

学号=INDIRECT(B\$10&"!I"&ROW(B3))

结果:

提取姓名(A10)个人所得税(B10)

姓名纳税人姓名

学号杨一二

分别获取"个人所得税"工作表 I2 单元格和 I3 单元格中的数据

3.结合其他函数更出彩

在 A2:A4 及 B2:B5 两个区域中求出大于等于 30 的个数

数据数据结果

20103

3010

4020

1550

=SUM(COUNTIF(INDIRECT({"A18:A21","B18:B21"}),">=30"))

利用 INDIRECT 函数将文本转化为引用的特殊性质构建内存数组

step1 COUNTIF 函数只能引用单一连续的单元格区域,使用 INDIRECT 函数欺骗 COUNTIF 达到各个区域分别求值形式内存数组的目的.

step2 再使用 SUM 函数对求出的结果进行求和,此时数组公式应使用 ctrl+shift+enter 结束.

二级组合框-indirect

作业:INDIRECT 函数的使用,数据验证,批量定义名称。

方法:

- 1.定义名称:以第1行文本命名名称,先按住 ctrl 选择这些长短不一的列,选择"定义名称"里的"根据所选内容创建",勾选"首行"
- 2.设置 A2 单元格数据验证: =省直辖市区,B2 的数据验证:=INDIRECT(A2),如图 8-4 所示;

在行政区划工作表中定义名称

- 1.名称定义:定义所有第1行名称
- 2.设置数据有效性
A2:=名称
A3:=INDIRECT(A2)

批量建立名称方法:

""""

使用 Ctrl 键帮助逐列选择(长度不一),然后选择"根据所选内容创建",在打开的对话框选择"首行"即可批量建立。

不带工作表名的名称定义:引用位置输入 :=!A1:A10

选中第一列 > 定义名称 > 名称:省直辖市区

选中其它所有市级的列 > 开始 > 编辑 > 查找和选择 > 定位条件 > 常量 > 公式 > 定义的名称 > 根据所选内容创建 > 只选中首行 > 名称管理器中就出现所有省市二级地区
""""

选择除台湾和香港的所有有数据的单元格区域 > 开始 > 查找与替换 > 定位条件 > 选择:常量, 勾选"数字,文本,逻辑值,错误."> 公式 > 定义的名称 > 根据所选内容创建 > 只选中"首行"

返回到二级组合框-indirect 中。

省直辖市区关联的市/区

定位到省直辖市区下面的单元格中 > 数据 > 数据工具 > 数据验证 > 全部清除 > 验证条件 > 允许:序列 > 来源:=省直辖市区

但这样生成的数据验证也会把行政区划中的标题省直辖市和自治区作为其中的一项,必须要修改,删除生成的标题.

定位到关联的市/区下面的单元格中 > 数据 > 数据工具 > 数据验证 > 允许:序列, 来源:=INDIRECT(\$A\$2)

这样,在关联的市区中就会使用 INDIRECT 间接引用从定义的名称中查找 A2 中的名称. 如果 A2 中是"北京市", INDIRECT 函数就会把"北京市"的引号去掉,转换为引用, 变为在名称管理器中定义的北京市的名称. 在关联的市区中显示出来.

这样就完成了二级关联引用.

INDEX 练习:

1 列变 5 列

A1:A100 中是 100 个人的名字, 把它们转换到 5 列 20 行中, 第一行中的数据为 1-5, 第二行中为 6-10, 如此类推

=INDIRECT("A"&5*ROW(A1)-5+COLUMN(A1))&"")

解释:

5*ROW(A1) -5+COLUMN(A1)

ROW(A1)的结果是 1, COLUMN(A1)的结果也是 1, 这里的公式相当于:

5*1-5+1, 计算结果是 1。

5*ROW(A1) -5+COLUMN(A1)的计算结果是 2。

当公式向下复制填充或是向右复制时, 就会看到变化了:

公式向右复制填充时, 因为在同行中, ROW(A1)的值不变, COLUMN(A1)依次变为 COLUMN(B1), COLUMN(C1).....结果就是 2, 3,这样的, 也就是相当于 5-5+2, 5-5+3.....

将生成以 1 递增的序列, 获取 A 列的行号数字。

在第一行中就得到了"A1", "A2", "A3", "A4", "A5"这样的值, 使用 INDIRECT 函数就转换为对单元格的引用, 进而得出 A1-A5 中的名字.

如果向下复制公式, ROW(A1)会依次变为 ROW(A2), ROW(A3).....结果就是 2, 3,这样的. 也就是相当于 5*2-5+1, 5*3-5+1.....

计算结果分别是 7, 12....., 即生成以 5 递增的序列。

第一列中就得到了"A1", "A6", "A11", ...这样的值. 使用 INDIRECT 函数转换为引用 A1, A6, A11, ...单元格, 取其中的名字.

公式最后使用&"的目的是为了屏蔽无意义的 0 值。

初级水平用户

以 VLOOKUP 函数来说, 你得会此第四个参为" TRUE" 的用法, 即模糊 查找用法; 有时候 VLOOKUP 函数明明写的是对的, 结果值就#N/A , 得掌握 排除此类故障的方法, 比如得知道可能是"制表符"捣乱, 影响了 VLOOKUP 函数的结果;

以 IF 函数来说, 得知道 IF 函数和 AND 函数, OR 函数的组合用法, 同时知道多个 IF 函数连续嵌套表达的技巧;

以 SUMIFS 函数来说, 得知道此中"条件"的多种表达方式, 知道用"引 用","&","通配符"等多种形式来表达复杂的求和条件

图片链接

水果名称	点击浏览照片
菠萝	菠萝
西瓜	西瓜
哈密瓜	哈密瓜
猕猴桃	猕猴桃
葡萄	葡萄
木瓜	木瓜
桃子	桃子
无花果	无花果
柿子	柿子
草莓	草莓

=HYPERLINK("D:\水果图片"&"\"&A2&".jpg",A2)

选中图片所在文件夹, 按住 shift, 右击, 选择复制为路径命令

在 B2 单元格中输入函数, 类似于=HYPERLINK("D:\水果图片\"&A2&".jpg",A2)

第一个参数, 拼接 A 列中对应的水果名称的图片所在的路径, 第二个参数, 显示什么名称.

数组

什么是数组？

数组是具有某种联系的多个元素集合。班级里 50 个学生，这里如果班级是数组 50 个学生就是数组里的 50 个元素。当然，班级里的是可变以 20 个，可以是 30 个，也可以是 60 个。Excel 里，班级就相当于工作表，而学生的单元格数据。

Excel 数组，理解是为多个单元格值的数组，理解是为多个单元格值的数组，理解是为多个单元格值的集合。

数组，理解是为多个单元格值的集合。

主要解决的问题是：快速处理多行列数据

数组大小要求

- 一维数组(单行或列)
 - 同维度(数组元素一致)
 - 不同维度的一数组，例如 1×3 与 5×1 ，结果是 5×3
- 二维数组(多行或列)与一维数组计算
 - 二者相同维度(行或列)的元素个数一致，否则出现 #N/A 错误
- 二维数组与计算
 - 行，列的元素个数一致，否则出现 #N/A 错误

数组公式

- 源数据：多个单元格构成
- 结果：
- 返回多个单元格 数组公式是相对于普通而言的。
 - 普通公式(如上面的 $=\text{SUM}(\text{B2:D2})$, $=\text{B2}+\text{C2}+\text{D2}$ 等)，只占用一个单元格，只返回一个结果。
 - 数组公式可以占用一个单元格，也可以占用多个单元格。它对一组数或多组数进行多重计算，并返回一个或多个结果
- 例如，函数 $\text{FREQUENCY}()$
- 返回单个元格

数组公式

$\{=\text{SMALL}(\text{IF}(\text{A1:B4}>0,\text{A1:B4}),\{1,2,3\})\}$

数组公式标志

- 在 Excel 中数组公式的显示是用大括号对 "{}" 来括住以区分于普通 Excel 公式。
- 数组常量的大括号可以手工输入

例: = {1,2,3}, {1,2}, {15,18}

- 数组公式必须使用 **Ctrl+Shift+Enter** 结束公式的输入。

文本必须用""括起来。

数组维数

用行数 (高)和列数 (宽)确定的数据矩形

一维数组: 单独的一行或列

二维数组: 多行多列

例: 单元格 A1:D3, 一个三行四列的二维数组

二维数组的间隔符号

间隔行 ; 间隔列 ,

水平数组 (1 行 5 列), 跨了 5 列, 使用列分隔符

- {1,2,3,4,5}, COLUMN(A:E)
- 垂直数组 (5 行 1 列), 跨了 5 行, 使用行分隔符
- {1; 2; 3; 4; 5}, ROW(1:5)
- 单元素数组
- {1}, row(1:1), column(A:A)
- {=SMALL(IF(A1:B4>0,B4),{1,2,3})}

规律

- 多组数进行计算, 返回一个结果
- 对应位置进行算术运
- 对应位置进行逻辑运算
- 再求和

- 多组数进行计算, 返回个结果
- 对应位置进行算术运(+, -, *, /)
- 对应位置进行逻辑运算(*, +)

数组优点

- 无需考虑公式若复制时设置地址引用的相对还是绝对
- 有时，无需复制公式， 就可以直接获取多个结果单元格
- 批量处理数据，效率高
- 逻辑关系清晰

运算符*和+ 代替逻辑函数 AND 和 OR

- 在普通函数中可以使用星号 * 代替 "与" (数组公式不行)

例: =IF(AND(A1>60,A1<=70), " 符合 ", " 不符合 ")

等价于:

=IF((A1>60)*(A1<=70), "符合","不符合 ")

- 加号 + 代替"或" (数组公式不行)

例 1: =OR(AND(B2="男",C2>60),AND(B2="女",C2>55))

等价于:

=AND(B2="男",C2>60) +AND(B2="女",C2>55)

例 2: =SUMPRODUCT((B2:B11={"江西","广东 "})*(C2:C11="男")*D2:D11)

=SUMPRODUCT(((B2:B11=" 江西 ") + (B2:B11=" 广东 "))* (C2:C11=" 男")*D2:D11)

- 双重运算符
- 取决于运算符两端操作数，若是逻辑值，实现逻辑与或运算；若一端或两端是数据，算术运算

在数组中不能用 AND, OR 代替 *, +

原因:数组公式需要执行多重计算，而 AND, OR 返回的是单值 TRUE 或 FALSE，不能形成数组公式多区域之间的一对应关系。

例: =SUM(AND(C3:C7>70, C3:C7<80)*1)}

而 AND({TRUE;FALSE;TRUE;TRUE;TRUE},{TRUE;TRUE;FALSE;TRUE;FALSE})返回是 FALSE

则=SUM(FALSE*1)=SUM(0*1)=0

正确:

$(C3:C7>70)*(C7<80)=\{TRUE; FALSE; TRUE; TRUE; TRUE\}*\{TRUE; TRUE; FALSE; TRUE; FALSE\}=\{1;0;1;1;1\}*\{1;1;0;1;0\}=\{1;0;0;1;0\}$
 $=SUM(\{1;0;0;1;0\})=2$

视频演示

计算多种商品的销售总金额

品名	销售数量	销售单价	小计
商品 1	100	¥20.00	¥ 2,000.00
商品 2	200	¥15.00	¥ 3,000.00
商品 3	300	¥38.00	¥ 11,400.00
商品 4	107	¥ 191.00	¥ 20,437.00
商品 5	858	¥ 61.00	¥ 52,338.00
商品 6	402	¥ 34.00	¥ 13,668.00
商品 7	654	¥ 374.00	¥ 244,596.00
商品 8	126	¥ 434.00	¥ 54,684.00

普通方法:

方法 1: 插入辅助列,先求出各商品的销售额,然后再求总和。

方法 2: 直接输入类似公式 $=SUM(B2*C2,B3*C3,B4*C4)$,效率低,实现费时。尤其有 100 行数据,一千行号数据呢?

数组方法:

方法 3: 输入公式 $=SUM(B2:B41*C2:C41)$,按 CTRL+SHIFT+ENTER 结束公式输入。

两列数据之差的和

作业:计算 B, A 数据之差的和。

A	B	B-A
1	100	
2	200	
3	300	

4	400	
5	500	
6	600	
7	700	
8	800	
9	900	
10	1000	

数组计算

=SUM(B2:B11-A2:A11)

使用 F9 来查看数组的计算.

先分别把 B2:B11 和 A2:A11 中的数据取出来, 成为两组数组常量, 再分别把数组常量中对应的值相减. 得到相减结果的一个数组常量, 然后再把这个数组常量中的元素相加.

判断身份证的位数是否合法

不同方法判断身份证输入位数是否正确, 即判断身份证号的长度是否为 15 位或 18 位.

姓名	身份证	一般方法	数组方法	数组常量
赵	511025198503196191	TRUE	TRUE	TRUE
钱	43250319881230435	FALSE	FALSE	FALSE
孙	511025770316628	TRUE	TRUE	TRUE
李	130301200308090514	TRUE	TRUE	TRUE
周	130502870529316	TRUE	TRUE	TRUE
吴	432503860923517	TRUE	TRUE	TRUE
郑	511022196802306112	TRUE	TRUE	TRUE
王	13030119990620515X	TRUE	TRUE	TRUE
冯	43250278065174	FALSE	FALSE	FALSE
陈	51102519770613517	FALSE	FALSE	FALSE

方法一: 一般方法

=OR(LEN(B2)=15,LEN(B2)=18)

方法二: 数组方法, 选择所有结果单元格区域

=IF((LEN(B2:B11)=15)+(LEN(B2:B11)=18),TRUE,FALSE)

说明:

1. 因为 LEN(B2:B11)=15 需要先计算出结果, 但逻辑运算的优先级低, 所以需要用()包起来.

2. +号的意义是, 如果发现两边是逻辑的 True 和 False 的话, 就做逻辑运算, 如果是数字的

话, 就使用算术运算. +在这里做逻辑或运算.

3. 这里不能用 AND , OR 代替 *, +

原因:

1. 数组公式需要执行多重计算, 而 AND, OR 返回的是单值 TRUE 或 FALSE, 不能形成数组公式

2. AND 中只有一个条件为 false, 结果就为 false, or 中只要有一个条件为 true, 结果就为 true. 这对于数组多行多列的计算是不行的.

方法三: 数组常量方法

=OR(LEN(B2)={15,18})

方法四, 分别判断长度是否为 15 位或 18 位, 然后再进行或运算.

选中与原数据列数相等的两列, 如 F2:G11 的单元格区域, 输入

=IF((LEN(B2:B11)={15,18}),TRUE,FALSE)

ctrl+shift+Enter 结束公式, 就会在 F2:F11 中得到 15 判断的结果, G2:G11 中得到 18 的判断结果.

使用 F9 查看运算过程

=IF({FALSE,TRUE;FALSE,FALSE;TRUE,FALSE;FALSE,TRUE;TRUE,FALSE;TRUE,FALSE;FALSE,TRUE;FALSE,TRUE;FALSE,FALSE;FALSE,FALSE},TRUE,FALSE)

其中的一对 FALSE,TRUE; 就分别对应着一行中对 15 和 18 判断的结果

如果在计算之前只选中了 F2:F11 的单元格区域, 那么就只会得到 15 的判断结果, 18 的判断结果就丢弃了.

再在第三列第一个单元格中输入 OR(F2,G11), 再填充到其它单元格中.

多条件求和: 统计一车间男职工工资总和。

姓名	部门	性别	工资
赵大年	一车间	男	2400
钱英姿	二车间	女	1700
孙军	一车间	男	2000
赵芳芳	一车间	女	1700
钱三金	三车间	男	1900
孙纹	一车间	女	1800
赵一曼	二车间	女	2000

钱芬芳	三车间	女	1700
孙大胜	三车间	男	2300

方法一：使用 SUM 函数来完成

定位到目标单元格中，输入

=SUM((B2:B10="一车间")*(C2:C10="男")*D2:D10)

虽然这里的目标单元格是一个，数组还要使用 ctrl+shift+Enter 的方式输入，否则就得不到正确的值。

这里的*进行逻辑与的运算。

前两个条件先进行运算，得到两组逻辑值的数组常量，把这两组数组常量进行逻辑与的运算，然后与工资这个数组常量进行一一相乘，然后把结果相加

方法二：使用 SUMPRODUCT 函数来完成

sumproduct 带条件的求和之王，在给定的几组数组中，将数组间对应的元素相乘，并返回乘积之和

=SUMPRODUCT((B2:B10="一车间")*(C2:C10="男")*D2:D10)

此函数是一个数组函数，只需要回车即可，而 SUM 函数是一个普通函数，如果想要用于数组，必须使用 ctrl+shift+回车

判断男女是否达到退休年龄

姓名	性别	年龄	一般方法	数组方法
赵	男	55	否	否
钱	女	46	否	否
孙	女	57	是	是
李	男	53	否	否
周	男	65	是	是
吴	女	53	否	否
郑	女	59	是	是
王	男	46	否	否
冯	男	60	是	是
陈	女	60	是	是

男士:60 岁及以上；女式:55 岁及以上。

AND/OR 方法

男,60 岁及以上

AND(B2="男",C2>=60)

女,55 岁及以上

AND(B2="女",C2>=55)

男,60 及以上或女,55 及以上

=OR(AND(B2="男",C2>=60),AND(B2="女",C2>=55))

退休与否

=IF(OR(AND(B2="男",C2>=60),AND(B2="女",C2>=55)),"是","否")

***/+替代 AND/OR 方法**

=IF(((B2="男")*(C2>=60))+((B2="女")*(C2>=55)),"是","否")

数组方法

选中所有结果单元格

=IF((B2:B11="男")*(C2:C11>=60)+(B2:B11="女")*(C2:C11>=55),"是","否")

ctrl+shift+Enter 结束输入

SUMPRODUCT 统计 3 天同品牌销售之和。

品牌	周一	周二	周三
西门子	39	75	38
长虹	85	34	17
Philip	10	71	80
TCL	90	53	47
西门子	75	58	55
西门子	13	87	60
长虹	78	50	75
TCL	17	22	30
长虹	50	58	73
Philip	33	98	15
TCL	82	17	10

品牌	3 天之和
西门子	
长虹	
Philip	
TCL	

复制品牌列表到新位置 > 数据 > 数据工具 > 删除重复项

方法一

使用 SUMIF()和 SUMIFS()只能先求出某天的和,再求出 3 天之和,即只能对一系列数据求和,

不能对多列数据求和.

=SUMIF(\$A\$2:\$A\$12,\$F2,B\$2:B\$12)

西门子周日的值, 要向右拖动得出周二周三的值, 所以对行锁定 B\$2:B\$12, 又要向下拖动得出其它品牌的值, 所以对品牌锁定列\$F2, 向下拖动时行发生变化, 就得到其它品牌的值.

方法二, 使用 SUMPRODUCT()条件求和之王

A2:A12 品牌列表, F2 西门子, B2:D12 为周一到周三三天各品牌销售数据

SUMPRODUCT 函数本身就是数组函数, 所以在输入时不用先选择单元格的区域, 只需要定位到本门子的销售总和单元格中即可, 在输入完成后也不需要按 ctrl+shift+回车, 并且可以拖动填充柄到其它单元格中.

=SUMPRODUCT((\$A\$2:\$A\$12=F2)*\$B\$2:\$D\$12)

注意, 必须要对 A2:A12 和 B2:D12 使用绝对引用, 否则复制到其它品牌的时候, 这两组数据也会向下移动, 就会出错.

如何验证, ctrl 键选中西门子的所有值, 看状态栏中的值

使用 F9 理解 sumproduct 的求解过程

1. \$A\$2:\$A\$12=F2

先判断 A2:A12 品牌列中的品牌是否等于 F2 单元格中的西门子, 得到逻辑类型的列数组常量.

{TRUE;FALSE;FALSE;FALSE;TRUE;TRUE;FALSE;FALSE;FALSE;FALSE;FALSE}

2. \$B\$2:\$D\$12 为 3*11 的逻辑型二维数组常量.

{39,75,38;85,34,17;10,71,80;90,53,47;75,58,55;13,87,60;78,50,75;17,22,30;50,58,73;33,98,15;82,17,10}

3. (\$A\$2:\$A\$12=F2)*\$B\$2:\$D\$12 二者相乘得到 3*11 的数值型二维数组常量. 只有品牌是西门子的行才不为 0, 其余行的数值都为 0.

{39,75,38;0,0,0;0,0,0;0,0,0;75,58,55;13,87,60;0,0,0;0,0,0;0,0,0;0,0,0;0,0,0}

4. 再对数组中非零值进行相加

产量大于 100 的和

机台	产量
1#	117

2#	120
3#	117
4#	87
5#	83
6#	98
7#	115
8#	81
9#	111
10#	110

	产量大于 100 的合计	小于 100 或者大于 110 的合计
SUM 函数:		
SUMPRODUCT 函数:		
SUMIF(S) 函数:		

统计列标题要求的合计。

SUM 函数:

产量大于 100 的合计

=SUM((B2:B11>100)*B2:B11)

ctrl+shift+enter

小于 100 或者大于 110 的合计

=SUM(((B2:B11<100)+(B2:B11>110))*B2:B11)

ctrl+shift+enter

这里的+号表示或的关系,条件之一

SUMPRODUCT 函数:

产量大于 100 的合计

=SUMPRODUCT((B2:B11>100)*B2:B11)

B2:B11>100 表示求和条件, 后面是求和区域

小于 100 或者大于 110 的合计

=SUMPRODUCT(((B2:B11<100)+(B2:B11>110))*B2:B11)

SUMIF(S)函数:

产量大于 100 的合计

=SUMIF(B2:B11,">100",B2:B11)

=SUMIFS(B2:B11,B2:B11,">100")

小于 100 或者大于 110 的合计

因为 SUMIF(S)函数本身就是处理与的关系的, 或的关系不能使用 SUMIFS 函数来实现.

如果要求求和的列和条件的列是同一列的话, 就可以省略求和的列

=SUMIF(B2:B11,">100")

SUMIF COUNTIF AVERAGEIF 中只要用到了关系运算符,就需要加上"",但如果是等于 100,就不需要加上引号,只要写成如下形式即可

=SUMIF(B2:B11,100,B2:B11)

SMALL 和 LARGE 函数

把每一行的数据各自进行降序排序。

2	3	4	7	2	2	3
3	4	2	2	6	3	2
5	5	8	2	3	8	2
5	6	8	9	4	5	6
5	6	2	5	7	2	5
3	8	1	4	9	3	5
1	9	4	4	5	4	1
1	9	4	4	5	1	4

同时调整多列的宽度: 同时选中多个列, 开始 > 单元格 > 格式 > 列宽

使用 SMALL 函数来获取某区域内第几小的数.

使用 LARGE 函数来找出某区域内第几大的数.

=LARGE(A2:A9,2), A2:A9 表示区域, 2 表示找出第 2 大的数.

选中与第一行数字个数相同的连续单元格 > =LARGE(A2:G2,{1,2,3,4,5,6,7}) > CTRL+SHIFT+ENTER

LARGE 的第二个参数是一个数组常量, 当数据量很大的时候, 再使用这种方法就很难

=ROW()函数能取到当前的行号

=COLUMN()能取到当前的列号

可以使用 ROW 和 COLUMN 函数来辅助产生任意有规律的数组.

求 1-10 各个数的和:

=SUM(ROW(1:10))

=LARGE(A2:G2,COLUMN()-12), 当前列是 13 列, 减去 12 就是 1,与前面的结果相同.

使用 SMALL 函数来完成

=SMALL(A2:G2,20-COLUMN())

进阶, 同时对多行多列的数据进行排序, 把排序的结果填充到与之大小相同的单元格区域中.

用随机函数产生原始数据，=RANDBETWEEN(1,99)

=SMALL(INDEX(A2:I10,,COLUMN(A:I)),ROW(1:9))

进行列升序排列。

用多单元格数组公式：

=SMALL(INDEX(A2:I10,ROW(1:9),),COLUMN(A:I))进行行升序排列。注意：公式为多单元格数组公式，一是需选定要输入的单元格区域，二要用数组的三键结束。多单元格数组公式的使用方法： 1、选定方法，直接在名称框中输入。比如你要选定 15*10 的区域，就直接在名称框中输入 A1:J15，回车后鼠标打击编辑栏，接着输入公式，完成后按数组的三键“ctrl+shift+enter”即可完成。

数组函数 FREQUENCY

以一系列垂直数组返回某个区域中数据的频率分布。例如，使用函数 FREQUENCY 可以计算在给定的分数范围内测验分数的个数。由于函数 FREQUENCY 返回一个数组，所以必须以数组公式的形式输入 (Ctrl+Shift+Enter)。

语法: FREQUENCY(data_array,bins_array)

Data_array 为一数组或对值的引用,来计算频率。

Bins_array 为间隔的数组或对间隔的引用,该间隔用于对 data_array 中的数值进行分组。

示例:

结果个数比分段点个数多 1

选择单元格 D2:D5，输入公式后按 Ctrl+Shift+Enter

ABDE

1 分数分段点公式结果

27970=FREQUENCY(A2:A10,B2:B4)1

38579=FREQUENCY(A2:A10,B2:B4)2

47889=FREQUENCY(A2:A10,B2:B4)4

585=FREQUENCY(A2:A10,B2:B4)2

650

781

895

988

1097

课后作业 1-平均分高于 90 的少数民族

要求:平均分高于 90 的少数民族。

姓名 平均成绩 民族 是否符合条件

赵 87 彝族

钱 72 俄罗斯族

孙 83 汉族

李 96 高山族

周 80 高山族

吴 76 汉族

郑 72 汉族

王 98 俄罗斯族

冯 71 俄罗斯族

陈 94 俄罗斯族

是否符合条件(一般方法)

=AND(B2>90,C2<>"汉族")

数组方法 1:

使用"*"

=(B2:B11>90)*(C2:C11<>"汉族")

数组方法 2:

使用"*"+IF

思考:

AND 使用到数组,将产生什么结果?

课后作业 2-前三名产量和

机台产量

1#1108

2#1185

3#1101

4#1068

5#1036

6#1000

7#935

8#955

9#1162

提示:使用 LARGE+SUM,数组常量。
数组方法实现,结果:3455

课后作业 3-姓赵

姓名 部门 性别 工资
赵大年 一车间 男 2400
钱英姿 二车间 女 1700
孙军 一车间 男 2000
赵芳芳 一车间 女 1700
钱三金 三车间 男 1900
孙纹 一车间 女 1800
赵一曼 二车间 女 2000
钱芬芳 三车间 女 1700
孙大胜 三车间 男 2300

姓赵的女职工工资总和

姓名用 LEFT 函数取第一个,如果等于赵.

进价-多条件求和

作业:利用数组实现如下多条件的计算。

客户	月份	销售额
A	1	100
B	2	200
C	3	200
C	4	200
A	1	200
C	2	200
A	3	200
A	4	200
B	1	200

项目	SUMIFS:	SUMPRODUCT:
----	---------	-------------

客户 A 销售额	700	700
客户 A 的 1 月份销售额	300	300
客户 A 的 1 月份和 3 月份销售额	500	500
客户 A 和 B 销售额	1100	1100
客户 A 和客户 C 的 1 月份销售额	300	300
客户 A 的 1 月份和客户 C 的 3 月份销售额	500	500
客户 A 和客户 C 的 1 月份和 3 月份销售额	700	700
客户 A 和客户 C 的 1 月/3 月份/4 月份销售额	1100	1100
客户 A/B/C 的 1 月/3 月份/4 月份销售额	1300	1300

客户 A 销售额

=SUMIF(A2:A10,"A",C2:C10)

=SUMIFS(C2:C10,A2:A10,"A")

=SUMPRODUCT((A2:A10="A")*C2:C10)

客户 A 的 1 月份销售额

=SUMIFS(C2:C10,A2:A10,"A",B2:B10,1)

=SUMPRODUCT((A2:A10="A")*(B2:B10=1)*C2:C10)

客户 A 的 1 月份和 3 月份销售额

=SUM(SUMIFS(C2:C10,A2:A10,"A",B2:B10,{1,3}))

=SUMPRODUCT((\$A\$2:\$A\$10="A")*(\$B\$2:\$B\$10={1,3})*\$C\$2:\$C\$10)

客户 A 和 B 销售额

=SUM(SUMIFS(C2:C10,A2:A10,{"A","B"}))

=SUMPRODUCT((\$A\$2:\$A\$10={"A","B"})*\$C\$2:\$C\$10)

客户 A 和客户 C 的 1 月份销售额

=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,1))

=SUMPRODUCT((\$A\$2:\$A\$10={"A","C"})*(B2:B10=1)*\$C\$2:\$C\$10)

客户 A 的 1 月份和客户 C 的 3 月份销售额

=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{1,3}))

=SUMPRODUCT((((\$A\$2:\$A\$10={"A"})*(\$B\$2:\$B\$10=1))+(\$A\$2:\$A\$10={"C"})*(\$B\$2:\$B\$10=3))*\$C\$2:\$C\$10)

客户 A 和客户 C 的 1 月份和 3 月份销售额

=SUM(SUMIFS(C1:C9,A1:A9,{"A","C"},B1:B9,{1})+SUMIFS(C1:C9,A1:A9,{"A","C"},B1:B9,{3}))

=SUMPRODUCT((((\$A\$2:\$A\$10={"A"})*(\$B\$2:\$B\$10={1,3}))+(\$A\$2:\$A\$10={"C"})*(\$B\$2:\$B\$10={1,3}))*\$C\$2:\$C\$10)

客户 A 和客户 C 的 1 月/3 月份/4 月份销售额

=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{1})+SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{3})+SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{4}))

=SUMPRODUCT((((\$A\$2:\$A\$10={"A"})*(\$B\$2:\$B\$10={1,3,4}))+(\$A\$2:\$A\$10={"C"})*(\$B\$2:\$B\$10={1,3,4}))*\$C\$2:\$C\$10)

客户 A/B/C 的 1 月/3 月份/4 月份销售额

=SUM(SUMIFS(C2:C10,A2:A10,{"A","B","C"},B2:B10,{1}))+SUMIFS(C2:C10,A2:A10,{"A","B","C"},B2:B10,{3}))+SUMIFS(C2:C10,A2:A10,{"A","B","C"},B2:B10,{4}))
=SUMPRODUCT((((\$A\$2:\$A\$10={"A"})*(\$B\$2:\$B\$10={1,3,4}))+(\$A\$2:\$A\$10={"B"})*(\$B\$2:\$B\$10={1,3,4}))+(\$A\$2:\$A\$10={"C"})*(\$B\$2:\$B\$10={1,3,4}))*\$C\$2:\$C\$10)

=SUMIFS(C2:C10,A2:A10,"A")
=SUMIFS(C2:C10,A2:A10,"A",B2:B10,1)
=SUM(SUMIFS(C2:C10,A2:A10,"A",B2:B10,{1,3}))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"}))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,1))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{1,3}))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{1,3}))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","C"},B2:B10,{1,3,4}))
=SUM(SUMIFS(C2:C10,A2:A10,{"A","B","C"},B2:B10,{1,3,4}))

图表

1. 图表建立与设置

选择合适的图表类型，以及图表类型的更换和组合

设置图表数据源，追加数据，已达到灵活组织图表的数据源

添加和格式化图表元素，以设置精美完善的图表

设置迷你图，条件格式，以微型化图表到单元格式里

设置可变的动态图表，用于经济领域商务图表，工程图表等，以提升图表的品质

图表

(1)图表类型

条形图 在水平方向上比较不同类型的数据

柱形图 在竖直方向上比较不同类型的数据

折线图 按类别显示一段时间内数据的变化趋势

饼图 在单组中，描述部分与整体的关系

XY 散点图 描绘两种相关数据的关系

面积图 强调一段时间内数值的相对重要性

圆环图 以一个或多个数据类别来比较部分与整体的关系

雷达图 表明数据或数据频率相对于中心的变化

曲面图 当第三个变量变化时，描绘另外两个变量的变化轨迹

气泡图 突出显示值的聚合，类似散点图

股价图 综合了柱形图和折线图,专门用来反映股票价格变化

曲面图

曲面图主要是表现 $z = f'(f(x), f(y))$ 这样的计算式

曲面图和 "模拟运算表 " 可谓是天生的一对 , 地造的一双 . 为什么 ? 曲面图和模拟运算表在实际工作中 , 遭遇到的概率小于万分之一 .

2. 图表组成元素

- (1)图表区
- (2)绘图区
- (3)图表标题
- (4)数据标签
- (5)数据系列
- (6)坐标轴
- (7)网格线
- (8)分类名称
- (9)图例
- (10)图表数据系列名称

3. 创建图表

使用"插入|图表"组创建图表

使用一次按钮创建图表

Alt+F1

F11

4. 技巧和方法

制作函数 图表

类型组合 图表

背景

坐标轴

图表高层次理解

1. 图表表达是一种诉求信息，直接从图表即可获取有着 明确，好理解的内容
 2. 图表是在讲述一个故事，读者乐意听的故事
 3. 图表方向:传递和述一种信息
- 图表是一种视觉语言

要求

清晰可见的图表势能

—— 数据点，数据间差异，数据趋势变化

<http://www.excelhome.net/>

数据透视表

Power Query

数据|获取数据|启动查询编辑器

选择数据源 > 查询 > 勾选"使用|查询|创建/编辑查询"

文件|选项|自定义功能区|勾选"开发工具"

开发工具选项卡|COM 加载项|勾选三项"Power XXX for Excel"

插入选项卡看是否有"三维地图"和"Power View"

开始|文件|选项|自定义功能区|插入|新建组|重命名:Power View|确保 Power View 处于选中状态|在常用命令中选所有命令|找到 PowerView|添加|上下拖动修改位置

实现来自于 Access Excel 等数据源，多个工作表的构建透视，以构成广泛的应用领域
建立不同的布局方式，以获取不同分类汇总的报表
设置不同的值显示方式，多个汇总方式，实现数据透视表的刷新，复制，移动，分组等
建立数据透视表的计算字段和计算项

数据透视表

1. 数据透视表

数据透视表是一种可以快速汇总大量的交互式报，总结信息分析工具，快速比较统计数据，

综合了 Excel 中数据排序, 筛选, 分类汇总和数据分析的优点, 可以方便的调整布局, 分类汇总方式, 灵活地以多种不同形式展示数据的特征和明细。

分类汇总只能实现一个字段; 数据透视表可以实现多个字段的分类汇总。

总之合理运用数据透视表进行计算与分析, 能使许多复杂的问题简单化并且极大提高了工作效率

简单!

易操作!

变化 灵活!

2. 创建数据透视表

插入 | 表格 | 数据透视表

1) 设置布局: 压缩, 大纲, 表格

2) 更改汇总方式

3) 设置总计

4) 设置分类汇总位置

5) 格式化

布局

行, 列筛选器: X 轴, Y 轴, Z 轴

3. 值显示方式

方法: 选择类似"求和项: 单价", 右击, 选择"值字段设置"项, 打开的对话框选择"值显示方式"选项卡设置

百分比

差异百分比

其他项: 可以是列或行的字段项

4. 组合功能

日期: 年, 月, 季度, n 天

时间: 小, 分, 秒

方法: 选中要分组的列标题, 鼠标右键打开快捷菜单先择"创建组", 在其对话框中设置

5. 数据透视表计算

计算字段 —— 发生在列标题(字段名称)上

数据透视表工具 |" 分析"选项卡 |" 计算"组 |字段, 项目和集 |计算字段
输入字段名称, 公式(函数)

计算项 —— 发生在列数据里(例如, 性别的男女)

数据透视表工具 |" 分析"选项卡 |" 计算"组 |字段, 项目和集 |计算项
输入名称(类似: 2012 年增长率, 两个学期之差)

公式(类似:= 1-'2011'/'2012', =秋季学期 -春季学期)

6. 多个工作表数据

多数据区域生成透视表

在"快速访问工具栏"添加工具"数据透视表与图向导"

添加多个数据区域

布局

调整

7. 切片器

切片器就是一个筛选组件, 它包含一组按钮, 能够快速地筛选数据透视表中的数据

切片器通常要与创建的数据透视表相关联

点击"数据透视表工具"的插入切片器", 弹出切片器对话框

链接多个透视表:点击"切片器工具"的"报表连接"操作

清除筛选期功能:单击切片器右上角含有叉的按钮, 从而删除筛选结果。

删除切片器: Del 键。

8. 格式化(美化)透视表

9. 利用数据透视表转换二维为一维表

10. 刷新

11. 数据透视表选项

12. 显示详细信息

14. 数据透视表函数 GETPIVOTDATA

功能:返回存储在数据透视表中的数据。

语法:

GETPIVOTDATA(data_field, pivot_table, [field1, item1, field2, item2], ...)

Data_field 必需。 包含要检索的数据字段名称, 用引号。

Pivot_table 必需。 数据透视表中的任何单元格, 单元格区域 或命名区域的引用。此信息用于确定包含要检索的数据的数据透视表。

Field1 , Item1 , Field2, Item2 可选。 描述要检索的 数据的 1 到 126 个字段名称 对和项目名称。

数据透视表特点:

1. 几秒钟就能完成销售汇总
2. 10 秒钟完成按季度统计数据(创建组)
3. 10 秒钟完成批量生多个销售报表
4. 批量生成多个 报表(按部门, 项目, 月份等)
5. 快速统计每个人的总量, 平均和最大值(值, 3 次, 改变汇总方式)
6. 计算销售占比:商品, 销售额, 店铺, 销售员等